

We apply *OreModules* to the example of a stirred tank. See H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972.

```
> with(Ore_algebra):
> with(OreModules):
```

## 1 Stirred tank: an ordinary differential model

### 1.1 Controllability, parametrizability, flatness, Brunovský canonical form

Let us consider the stirred tank studied in H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, p. 7-10. Let us define the ring of ordinary differential operators with polynomial coefficients and with constants  $\theta$ ,  $V0$ ,  $c0$ ,  $c1$ ,  $c2$ .

```
> Alg := DefineOreAlgebra(diff=[D,t], polynom=[t], comm=[theta,V0,c0,c1,c2]):
```

Let us enter the matrix  $R$  which defines the system.

```
> R := evalm([[D+1/(2*theta),0,-1,-1],[0,D+1/theta,-(c1-c0)/V0,-(c2-c0)/V0]]);
```

$$R := \begin{bmatrix} D + \frac{1}{2\theta} & 0 & -1 & -1 \\ 0 & D + \frac{1}{\theta} & -\frac{c1 - c0}{V0} & -\frac{c2 - c0}{V0} \end{bmatrix}$$

The equations of the system are then defined by:

```
> ApplyMatrix(R, [x1(t),x2(t),u1(t),u2(t)], Alg)=evalm([[0]$2]);
```

$$\begin{bmatrix} \frac{1}{2} \frac{x1(t)}{\theta} + \left(\frac{d}{dt} x1(t)\right) - u1(t) - u2(t) \\ \frac{x2(t)}{\theta} + \left(\frac{d}{dt} x2(t)\right) - \frac{(c1 - c0) u1(t)}{V0} - \frac{(c2 - c0) u2(t)}{V0} \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

Let us compute the formal adjoint of  $R$ .

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -D + \frac{1}{2\theta} & 0 \\ 0 & -D + \frac{1}{\theta} \\ -1 & -\frac{c1}{V0} + \frac{c0}{V0} \\ -1 & -\frac{c2}{V0} + \frac{c0}{V0} \end{bmatrix}$$

Now, let us check whether or not the stirred tank is parametrizable and, equivalently, flat. In order to do that, we need to compute the first extension module  $\text{ext}^1$  with values in  $Alg$  of the module associated with  $R\_adj$ .

```
> st := time(): Ext := Exti(R_adj, Alg, 1); time()-st;
```

$$\begin{aligned}
Ext := & \left[ \begin{array}{cc} 1 & 0 \\ 0 & 1 \end{array} \right], \left[ \begin{array}{cccc} 2D\theta + 1 & 0 & -2\theta & -2\theta \\ 0 & D\theta V0 + V0 & -c1\theta + \theta c0 & -c2\theta + \theta c0 \end{array} \right], \\
& [-2c1\theta V0 + 2c2\theta V0, -2c1\theta V0 + 2c2\theta V0] \\
& [-2c2c1\theta + 2c2^2\theta + 2c0c1\theta - 2c0c2\theta, \\
& -2c1^2\theta + 2c2c1\theta + 2c0c1\theta - 2c0c2\theta] \\
& [-c2V0 + c0V0, -2c1D\theta V0 + 2c2D\theta V0 - 2c1V0 + c2V0 + c0V0] \\
& [-c1V0 - 2c1D\theta V0 - c0V0 + 2c2D\theta V0 + 2c2V0, c1V0 - c0V0] \\
& 0.086
\end{aligned}$$

Therefore, we see that the stirred tank is generically parametrizable, and thus, generically flat. A generic parametrization of the system is given by  $Ext[3]$ . One can obtain this parametrization directly by using the command *Parametrization* as follows:

> `evalm([[x1(t)], [x2(t)], [u1(t)], [u2(t)]])=Parametrization(R, Alg);`

$$\begin{aligned}
& \begin{bmatrix} x1(t) \\ x2(t) \\ u1(t) \\ u2(t) \end{bmatrix} = \\
& [-2\xi_1(t)c1\theta V0 + 2\xi_1(t)c2\theta V0 - 2\xi_2(t)c1\theta V0 + 2\xi_2(t)c2\theta V0] \\
& \left[ -2\xi_1(t)c2c1\theta + 2\xi_1(t)c2^2\theta + 2\xi_1(t)c0c1\theta - 2\xi_1(t)c0c2\theta - 2\xi_2(t)c1^2\theta \right. \\
& \left. + 2\xi_2(t)c2c1\theta + 2\xi_2(t)c0c1\theta - 2\xi_2(t)c0c2\theta \right] \\
& \left[ V0(-\xi_1(t)c2 + \xi_1(t)c0 - 2\xi_2(t)c1 + \xi_2(t)c2 + \xi_2(t)c0 - 2\left(\frac{d}{dt}\xi_2(t)\right)c1\theta \right. \\
& \left. + 2\left(\frac{d}{dt}\xi_2(t)\right)c2\theta \right] \\
& \left[ -V0(\xi_1(t)c1 + \xi_1(t)c0 - 2\xi_1(t)c2 + 2\left(\frac{d}{dt}\xi_1(t)\right)c1\theta - 2\left(\frac{d}{dt}\xi_1(t)\right)c2\theta - \xi_2(t)c1 \right. \\
& \left. + \xi_2(t)c0) \right]
\end{aligned}$$

Let us find the relations of the coefficients for which the stirred tank becomes non-flat. We know that the stirred tank is flat if and only if the *Alg*-module  $M$  associated with the matrix  $R$  is free, i.e., projective as *Alg* is a left hereditary ring. But,  $R$  has full row rank as we can easily check by computing the *Alg*-linear relations of the rows of  $R$ , i.e., by computing the syzygy module of  $R$ :

> `SyzygyModule(R, Alg);`

INJ(2)

We find that there is no non-trivial relation on the rows of  $R$ , and thus,  $R$  has full row rank. Then, the *Alg*-module  $M$  is projective if and only if there exists a right-inverse of  $R$ . Let us check when such a right-inverse exists:

> `RightInverse(R, Alg);`

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \\ \frac{c2}{c1-c2} - \frac{c0}{c1-c2} & -\frac{V0}{c1-c2} \\ -\frac{c1}{c1-c2} + \frac{c0}{c1-c2} & \frac{V0}{c1-c2} \end{bmatrix}$$

Hence, a right-inverse of  $R$  exists if and only if  $c1 \neq c2$ . As the matrix  $R$  does not contain any polynomial coefficients (i.e. is independent of  $t$ ), we know that  $M$  is projective if and only if  $M$  is free. We can compute a generic flat output of the stirred tank by computing a left-inverse of the parametrization  $Ext[3]$ :

```
> S := map(factor, LeftInverse(Ext[3], Alg));
```

$$S := \begin{bmatrix} \frac{-c1 + c0}{2\theta V0 (c1 - c2)^2} & \frac{1}{2\theta (c1 - c2)^2} & 0 & 0 \\ -\frac{-c2 + c0}{2\theta V0 (c1 - c2)^2} & -\frac{1}{2\theta (c1 - c2)^2} & 0 & 0 \end{bmatrix}$$

Therefore, a flat output of the stirred tank is  $\xi = S (x1 : x2 : u1 : u2)^T$ , where  $\xi$  satisfies

$$(x1 : x2 : u1 : u2)^T = Ext[3] \xi.$$

Let us point out that we find again the condition  $c1 \neq c2$ . We have:

```
> evalm([[xi1(t)], [xi2(t)]] = ApplyMatrix(S, [x1(t), x2(t), u1(t), u2(t)], Alg);
```

$$\begin{bmatrix} \xi1(t) \\ \xi2(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2} \frac{(-c1 + c0) x1(t)}{\theta V0 (c1 - c2)^2} + \frac{1}{2} \frac{x2(t)}{\theta (c1 - c2)^2} \\ -\frac{1}{2} \frac{(-c2 + c0) x1(t)}{\theta V0 (c1 - c2)^2} - \frac{1}{2} \frac{x2(t)}{\theta (c1 - c2)^2} \end{bmatrix}$$

If we substitute  $\xi1$  and  $\xi2$  obtained above into the parametrization  $Ext[3]$ , we get the following matrix:

```
> Flat := simplify(Mult(Ext[3], S, Alg));
```

$$Flat := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ \frac{(2D\theta + 1)(-c2 + c0)}{2\theta (c1 - c2)} & \frac{(D\theta + 1)V0}{\theta (c1 - c2)} & 0 & 0 \\ -\frac{(2D\theta + 1)(-c1 + c0)}{2\theta (c1 - c2)} & -\frac{(D\theta + 1)V0}{\theta (c1 - c2)} & 0 & 0 \end{bmatrix}$$

This matrix shows that we can also use  $(x1 : x2)^T$  as a flat output of the system and we have the following parametrization:

```
> evalm([[x1(t)], [x2(t)], [u1(t)], [u2(t)]] = ApplyMatrix(Flat, [x1(t), x2(t), u1(t), u2(t)], Alg);
```

$$\begin{bmatrix} x1(t) \\ x2(t) \\ u1(t) \\ u2(t) \end{bmatrix} = \begin{bmatrix} x1(t) \\ x2(t) \\ \frac{1}{2} \frac{(-c2 + c0) x1(t)}{\theta (c1 - c2)} + \frac{1}{2} \frac{(-2c2 + 2c0) (\frac{d}{dt} x1(t))}{c1 - c2} + \frac{V0 x2(t)}{\theta (c1 - c2)} + \frac{V0 (\frac{d}{dt} x2(t))}{c1 - c2} \\ -\frac{1}{2} \frac{(-c1 + c0) x1(t)}{\theta (c1 - c2)} - \frac{1}{2} \frac{(-2c1 + 2c0) (\frac{d}{dt} x1(t))}{c1 - c2} - \frac{V0 x2(t)}{\theta (c1 - c2)} - \frac{V0 (\frac{d}{dt} x2(t))}{c1 - c2} \end{bmatrix}$$

Let us compute the Brunovsky canonical form for  $R$ :

```
> B := Brunovsky(R, Alg);
```

$$B := \begin{bmatrix} \frac{-c1 + c0}{2\theta V0 (c1 - c2)^2} & \frac{1}{2\theta (c1 - c2)^2} & 0 & 0 \\ -\frac{-c1 + c0}{4\theta^2 V0 (c1 - c2)^2} & -\frac{1}{2\theta^2 (c1 - c2)^2} & 0 & -\frac{1}{2\theta (c1 - c2) V0} \\ -\frac{-c2 + c0}{2\theta V0 (c1 - c2)^2} & -\frac{1}{2\theta (c1 - c2)^2} & 0 & 0 \\ \frac{-c2 + c0}{4\theta^2 V0 (c1 - c2)^2} & \frac{1}{2\theta^2 (c1 - c2)^2} & -\frac{1}{2\theta (c1 - c2) V0} & 0 \end{bmatrix}$$

Then, we obtain the following change of variables between the Brunovský variables  $\zeta[i]$  and the system variables  $x1, x2, u1$  and  $u2$ :

```
> evalm([[z1(t)], [v1(t)], [z2(t)], [v2(t)]])=ApplyMatrix(B,
> [x1(t), x2(t), u1(t), u2(t)], Alg);
```

$$\begin{bmatrix} z1(t) \\ v1(t) \\ z2(t) \\ v2(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2} \frac{(-c1 + c0) x1(t)}{\theta V0 (c1 - c2)^2} + \frac{1}{2} \frac{x2(t)}{\theta (c1 - c2)^2} \\ -\frac{1}{4} \frac{(-c1 + c0) x1(t)}{\theta^2 V0 (c1 - c2)^2} - \frac{1}{2} \frac{x2(t)}{\theta^2 (c1 - c2)^2} - \frac{1}{2} \frac{u2(t)}{\theta (c1 - c2) V0} \\ -\frac{1}{2} \frac{(-c2 + c0) x1(t)}{\theta V0 (c1 - c2)^2} - \frac{1}{2} \frac{x2(t)}{\theta (c1 - c2)^2} \\ \frac{1}{4} \frac{(-c2 + c0) x1(t)}{\theta^2 V0 (c1 - c2)^2} + \frac{1}{2} \frac{x2(t)}{\theta^2 (c1 - c2)^2} - \frac{1}{2} \frac{u1(t)}{\theta (c1 - c2) V0} \end{bmatrix}$$

Let us notice that the Brunovský canonical form exists if and only if  $c1 \neq c2$ .

Let us check that the new variables  $\zeta[i]$  satisfy the Brunovský canonical form. In order to do that, modulo the system equations, we need to eliminate the variables  $x1, x2, u1$  and  $u2$  from the previous system.

```
> F := Elimination(linalg[stackmatrix](B, R), [x1, x2, u1, u2],
> [z1, v1, z2, v2, 0, 0], Alg);
> ApplyMatrix(F[1], [x1(t), x2(t), u1(t), u2(t)], Alg)=ApplyMatrix(F[2],
> [z1(t), v1(t), z2(t), v2(t)], Alg);
```

$$\begin{bmatrix} 0 \\ 0 \\ u2(t) \\ u1(t) \\ x2(t) \\ x1(t) \end{bmatrix} = \begin{bmatrix} [-\frac{d}{dt} z2(t) + v2(t)] \\ [-\frac{d}{dt} z1(t) + v1(t)] \\ [(-c0 V0 - c1 V0 + 2 c2 V0) z1(t) + \%1 v1(t) + (c1 V0 - c0 V0) z2(t)] \\ [(-c2 V0 + c0 V0) z1(t) + (-2 c1 V0 + c0 V0 + c2 V0) z2(t) + \%1 v2(t)] \\ [(-2 c2 c1 \theta + 2 c2^2 \theta + 2 c0 c1 \theta - 2 c0 c2 \theta) z1(t) \\ + (-2 c1^2 \theta + 2 c2 c1 \theta + 2 c0 c1 \theta - 2 c0 c2 \theta) z2(t)] \\ [\%1 z1(t) + \%1 z2(t)] \\ \%1 := -2 c1 \theta V0 + 2 c2 \theta V0 \end{bmatrix}$$

The first two equations show that  $\zeta[i]$  satisfy the Brunovský canonical form.

Moreover, let us point out that the last equations give  $x1, x2, u1$  and  $u2$  in terms of  $z1, v1, z2$  and  $v2$ .

## 1.2 Autonomous elements, first integrals of motion

The concentrations  $c1$  and  $c2$  satisfy some linear equations with  $c0$  coming from the linearization of a non-linear model. For more details, see H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, p. 8.

In particular, from these linear equations, we obtain that  $c1 = c2$  implies  $c0 = c1 = c2$ . Hence, let us study the case where  $c2 = c1 = c0$ . The new system matrix becomes:

```
> R1mod := subs([c2=c0,c1=c0], evalm(R));
```

$$R1mod := \begin{bmatrix} D + \frac{1}{2\theta} & 0 & -1 & -1 \\ 0 & D + \frac{1}{\theta} & 0 & 0 \end{bmatrix}$$

Let us define the formal adjoint  $R1mod\_adj$  of  $R1mod$ .

```
> R1mod_adj := Involution(R1mod, Alg);
```

$$R1mod\_adj := \begin{bmatrix} -D + \frac{1}{2\theta} & 0 \\ 0 & -D + \frac{1}{\theta} \\ -1 & 0 \\ -1 & 0 \end{bmatrix}$$

We compute the first extension module with values in  $Alg$  of the  $Alg$ -module associated with  $R1mod\_adj$ :

```
> Ext1mod := Exti(Involution(R1mod, Alg), Alg, 1);
```

$$Ext1mod := \left[ \begin{bmatrix} 1 & 0 \\ 0 & D\theta + 1 \end{bmatrix}, \begin{bmatrix} 2D\theta + 1 & 0 & -2\theta & -2\theta \\ 0 & 1 & 0 & 0 \end{bmatrix}, \begin{bmatrix} -2\theta & 0 \\ 0 & 0 \\ 0 & 1 \\ -2D\theta - 1 & -1 \end{bmatrix} \right]$$

We obtain that there is a non-trivial torsion element in the  $Alg$ -module associated with  $R1mod$  defined by  $x2$ . It satisfies the equation  $(\theta D + 1)x2 = 0$ . Therefore, the system is non-controllable. We can find directly the autonomous elements in the system by using the following command:

```
> AutonomousElements(R1mod, [x1(t),x2(t),u1(t),u2(t)], Alg);
```

$$[[ \theta_1(t) + \theta \left(\frac{d}{dt} \theta_1(t)\right) = 0 ], [ \theta_1 = -C1 e^{(-\frac{t}{\theta})} ], [ \theta_1 = x2(t) ]]$$

The previous autonomous element corresponds to the following first integral of motion:

```
> V := FirstIntegral(R1mod, [x1(t),x2(t),u1(t),u2(t)], Alg);
```

$$V := -C1 e^{(\frac{t}{\theta})} x2(t)$$

By definition of the first integral of motion, the time derivative of  $V$  must be 0. Let us check it:

```
> Vdot := diff(V,t);
```

$$Vdot := \frac{-C1 e^{(\frac{t}{\theta})} x2(t)}{\theta} + -C1 e^{(\frac{t}{\theta})} \left(\frac{d}{dt} x2(t)\right)$$

The left hand side of the system equations is defined by:

> Sys := ApplyMatrix(R1mod, [x1(t), x2(t), u1(t), u2(t)], Alg);

$$Sys := \begin{bmatrix} \frac{1}{2} \frac{x1(t)}{\theta} + \left(\frac{d}{dt} x1(t)\right) - u1(t) - u2(t) \\ \frac{x2(t)}{\theta} + \left(\frac{d}{dt} x2(t)\right) \end{bmatrix}$$

We can extract the first derivative of  $x2$  in the second equation of  $Sys$ :

> sol := solve(Sys[2,1], diff(x2(t), t));

$$sol := -\frac{x2(t)}{\theta}$$

By substituting into  $Vdot$ , we finally obtain that the time derivative of  $V$  equals:

> subs(diff(x2(t), t)=sol, Vdot);

0

Therefore, we have checked that  $V$  is a first integral of motion. The fact that the system is not controllable is equivalent to the fact that the  $Alg$ -module associated with  $R1mod$  is not projective, i.e., to the fact that  $R1mod$  does not admit a right-inverse:

> RightInverse(R1mod, Alg);

□

A flat output of the controllable part, namely the following system

> Sys\_contr := ApplyMatrix(Ext1mod[2], [x1(t), x2(t), u1(t), u2(t)], Alg)=  
> evalm([[0], [0]]);

$$Sys\_contr := \begin{bmatrix} x1(t) + 2\theta \left(\frac{d}{dt} x1(t)\right) - 2\theta u1(t) - 2\theta u2(t) \\ x2(t) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

is obtained by computing a left-inverse of  $Ext1mod[3]$ :

> S2 := LeftInverse(Ext1mod[3], Alg);

$$S2 := \begin{bmatrix} -\frac{1}{2\theta} & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

Therefore, we have  $\xi = S (x1 : x2 : u1 : u2)^T$ , where  $\xi$  satisfies the parametrization of the controllable part, i.e.,  $(x1 : x2 : u1 : u2)^T = Ext1mod[3] \xi$ .

Finally, we can directly parametrize all solutions of  $R1mod (x1 : x2 : u1 : u2)^T = 0$  by computing:

> P1mod := Parametrization(R1mod, Alg);

$$P1mod := \begin{bmatrix} -2\theta \xi_1(t) \\ -C1 e^{-\frac{t}{\theta}} \\ \xi_2(t) \\ -\xi_1(t) - 2\theta \left(\frac{d}{dt} \xi_1(t)\right) - \xi_2(t) \end{bmatrix}$$

In particular, we can easily check that  $P1mod$  parametrizes a part of the solution as we have:

> ApplyMatrix(R1mod, P1mod, Alg);

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

In fact, we can prove that we parametrize all the  $C^\infty$  solutions. For more details, we refer the reader to A. Quadrat, D. Robertz, "On Monge problem for uncontrollable linear systems", to appear.

Let us show how to find again some of the previous about controllability by using the standard techniques for Kalman systems.

First of all, let us introduce the Kalman system  $Dx = Fx + Gu$ , where the matrices  $F$  and  $G$  are defined by:

```
> F := evalm([-1/(2*theta),0],[0,-1/theta]);
```

$$F := \begin{bmatrix} -\frac{1}{2\theta} & 0 \\ 0 & -\frac{1}{\theta} \end{bmatrix}$$

```
> G := evalm([[1,1],[(c1-c0)/V0,(c2-c0)/V0]]);
```

$$G := \begin{bmatrix} 1 & 1 \\ \frac{c1-c0}{V0} & \frac{c2-c0}{V0} \end{bmatrix}$$

Let us compute the controllability matrix of the system:

```
> C := ControllabilityMatrix(F, G, 2, Alg);
```

$$C := \begin{bmatrix} 1 & 1 & -\frac{1}{2\theta} & -\frac{1}{2\theta} \\ \frac{c1-c0}{V0} & \frac{c2-c0}{V0} & \frac{-c1+c0}{\theta V0} & \frac{-c2+c0}{\theta V0} \end{bmatrix}$$

```
> l := combinat[choose]([1,2,3,4], 2);
```

$$l := [[1, 2], [1, 3], [1, 4], [2, 3], [2, 4], [3, 4]]$$

```
> Sing := {seq(numer(linalg[det](linalg[submatrix](C, 1..2, l[i]))) , i=1..nops(l))};
```

$$Sing := \{-c2 + c0, -c1 + c0, -2c2 + c0 + c1, -2c1 + c0 + c2, -c1 + c2\}$$

```
> solve(Sing);
```

$$\{c0 = c2, c1 = c2, c2 = c2\}$$

Therefore, the controllability matrix  $C$  has rank 1 and thus, the system is not controllable, if and only if we have  $c0 = c1 = c2$ .

### 1.3 Observability, input-output behaviour

Now, let us introduce the following outputs of the system  $y1 = x1/(2\theta)$  and  $y2 = x2$ . See H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, p. 9. Let us study the observability of the system as well as the input-output behaviour.

```
> Ry := evalm([[1/(2*theta),0],[0,1]]);
```

$$Ry := \begin{bmatrix} \frac{1}{2\theta} & 0 \\ 0 & 1 \end{bmatrix}$$

Let us consider the matrix  $Rf = (R^T : (Ry : 0)^T)^T$ , namely:

```
> Rf := linalg[stackmatrix](R, linalg[augment](Ry, matrix(2,2,0)));
```

$$Rf := \begin{bmatrix} D + \frac{1}{2\theta} & 0 & -1 & -1 \\ 0 & D + \frac{1}{\theta} & -\frac{c1 - c\theta}{V\theta} & -\frac{c2 - c\theta}{V\theta} \\ \frac{1}{2\theta} & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix}$$

Now, we consider the system of equations  $Rf (x1 : x2 : u1 : u2)^T = (0 : 0 : y1 : y2)^T$  and we try to eliminate the state variables  $x1$  and  $x2$ . We obtain:

$$\begin{aligned} &> E := \text{Elimination}(Rf, [x1,x2,u1,u2], [0,0,y1,y2], Alg, [u1,u2]); \\ E &:= \text{table}([1 = \begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 0 & 1 \\ 1 & 0 \end{bmatrix}, 2 = \begin{bmatrix} 0 & -D\theta V\theta - V\theta & c1\theta - \theta c\theta & c2\theta - \theta c\theta \\ -2D\theta - 1 & 0 & 1 & 1 \\ 0 & 1 & 0 & 0 \\ 2\theta & 0 & 0 & 0 \end{bmatrix}]) \end{aligned}$$

Equivalently, in terms of the system variables, we have:

$$\begin{aligned} &> \text{ApplyMatrix}(E[1], [x1(t),x2(t)], Alg) = \text{ApplyMatrix}(E[2], \\ &> [y1(t),y2(t),u1(t),u2(t)], Alg); \\ \begin{bmatrix} 0 \\ 0 \\ x2(t) \\ x1(t) \end{bmatrix} &= \begin{bmatrix} -V\theta y2(t) - \theta V\theta (\frac{d}{dt} y2(t)) + (c1\theta - \theta c\theta) u1(t) + (c2\theta - \theta c\theta) u2(t) \\ -y1(t) - 2\theta (\frac{d}{dt} y1(t)) + u1(t) + u2(t) \\ y2(t) \\ 2\theta y1(t) \end{bmatrix} \end{aligned}$$

In particular, we find that the state variables  $x1$  and  $x2$  can be expressed without derivatives in terms of  $y1$ ,  $y2$ ,  $u1$  and  $u2$ . Therefore, the system defined by  $Rf$  is observable. Moreover, the first two previous equations give the input-output behaviour of the system (elimination of the state variables), i.e., we have:

$$\begin{aligned} &> IO := \{\text{seq}(\text{ApplyMatrix}(\text{linalg}[\text{submatrix}](E[2], 1..2, 1..4), \\ &> [y1(t), y2(t), u1(t), u2(t)], Alg) [i, 1] = 0, i = 1..2)\}; \\ IO &:= \{-V\theta y2(t) - \theta V\theta (\frac{d}{dt} y2(t)) + (c1\theta - \theta c\theta) u1(t) + (c2\theta - \theta c\theta) u2(t) = 0, \\ &-y1(t) - 2\theta (\frac{d}{dt} y1(t)) + u1(t) + u2(t) = 0\} \end{aligned}$$

To finish, let us point out that there exists another way to check whether or not the previous system is observable: we first write the system as  $Ro (x1 : x2)^T = Ryu (u1 : u2 : y1 : y2)^T$  and we check whether or not there exists a left-inverse of the matrix  $Ro = ((D I - A)^T : C^T)^T$ .

$$\begin{aligned} &> R0 := \text{linalg}[\text{submatrix}](Rf, 1..4, 1..2); \\ R0 &:= \begin{bmatrix} D + \frac{1}{2\theta} & 0 \\ 0 & D + \frac{1}{\theta} \\ \frac{1}{2\theta} & 0 \\ 0 & 1 \end{bmatrix} \end{aligned}$$

The system can be now written as  $Ro (x1 : x2)^T = Ryu (u1 : u2 : y1 : y2)^T$ , where  $Ryu$  is defined by:



```
> Ryu := linalg[diag](linalg[submatrix](-Rf, 1..2, 3..4),
> linalg[band]([1],2));
```

$$Ryu := \begin{bmatrix} 1 & 1 & 0 & 0 \\ \frac{c1 - c0}{V0} & \frac{c2 - c0}{V0} & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> S0 := LeftInverse(R0, Alg);
```

$$S0 := \begin{bmatrix} 0 & 0 & 2\theta & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Therefore, a left-inverse of  $R0$  exists, and thus, the system is observable.

```
> Rxyu := Mult(S0, Ryu, Alg);
```

$$Rxyu := \begin{bmatrix} 0 & 0 & 2\theta & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Finally, we recover the following expressions:

```
> evalm([[x1(t)], [x2(t)]])=ApplyMatrix(Rxyu, [u1(t), u2(t), y1(t), y2(t)], Alg);
```

$$\begin{bmatrix} x1(t) \\ x2(t) \end{bmatrix} = \begin{bmatrix} 2\theta y1(t) \\ y2(t) \end{bmatrix}$$

## 1.4 Motion planning, tracking

Let us recall that if  $c1 \neq c2$ , then we saw that the system is flat, i.e., parametrizable by means of an injective parametrization.


Moreover, we computed the parametrization  $(x1 : x2 : u1 : u2)^T = P (x1 : x2)^T$  of the system variables in terms of the flat outputs  $x1$  and  $x2$  of the system, where  $P$  is defined by:

```
> P := ApplyMatrix(Flat, [x1(t), x2(t), u1(t), u2(t)], Alg);
```

$$P := \begin{bmatrix} x1(t) \\ x2(t) \\ \frac{1}{2} \frac{(-c2 + c0)x1(t)}{\theta(c1 - c2)} + \frac{1}{2} \frac{(-2c2 + 2c0)(\frac{d}{dt}x1(t))}{c1 - c2} + \frac{V0x2(t)}{\theta(c1 - c2)} + \frac{V0(\frac{d}{dt}x2(t))}{c1 - c2} \\ -\frac{1}{2} \frac{(-c1 + c0)x1(t)}{\theta(c1 - c2)} - \frac{1}{2} \frac{(-2c1 + 2c0)(\frac{d}{dt}x1(t))}{c1 - c2} - \frac{V0x2(t)}{\theta(c1 - c2)} - \frac{V0(\frac{d}{dt}x2(t))}{c1 - c2} \end{bmatrix}$$

In particular, if we give some references trajectories  $x1ref$  and  $x2ref$ , then we explicitly get the corresponding open-loop control  $u1ref$  and  $u2ref$ . For instance, if we consider the following desired trajectories

```
> x1ref := 0.1*exp(-alpha*t): x2ref := 0.2*exp(-beta*t):
> alpha := 1/10: beta := 1/7:
> plot([x1ref, x2ref], t=0..70, color=[red, green]);
```




then, the corresponding open-loop inputs  $u1ref$  and  $u2ref$  are defined by:

```

> u1ref := subs(c1=1,c2=2,c0=1.25,V0=1,theta=50,
> ApplyMatrix(linalg[submatrix](Flat, 3..4, 1..2), [x1ref,x2ref], Alg)[1,1]);
      u1ref := -0.006750000000 e(-t/10) + 0.02457142857 e(-t/7)
> u2ref := subs(c1=1,c2=2,c0=1.25,V0=1,theta=50,
> ApplyMatrix(linalg[submatrix](Flat, 3..4, 1..2), [x1ref,x2ref], Alg)[2,1]);
      u2ref := -0.002250000000 e(-t/10) - 0.02457142857 e(-t/7)
> plot([u1ref,u2ref], t=0..70, color=[blue,magenta]);

```




Using the values given in page 10 of H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, we can numerically simulate the system for the given  $u1ref$  and  $u2ref$ . We obtain:

```
> ODE1 := {seq(ApplyMatrix(subs([c1=1,c2=2,c0=1.25,V0=1,theta=50], evalm(R)),
> [x1(t),x2(t),u1ref,u2ref], Alg)[i,1]=0,i=1..2)};
```

$$ODE1 := \left\{ \frac{1}{50} x_2(t) + \left( \frac{d}{dt} x_2(t) \right) + 0.02457142857 e^{(-\frac{t}{7})} = 0, \right.$$

$$\left. \frac{1}{100} x_1(t) + \left( \frac{d}{dt} x_1(t) \right) + 0.009000000000 e^{(-\frac{t}{10})} = 0 \right\}$$

```
> IC := {x1(0)=0.1, x2(0)=0.2}:
> dsol1 := dsolve(ODE1 union IC, {x1(t),x2(t)}, type=numeric, stiff=true):
> plots[odeplot](dsol1, [[t,x1(t), color=blue], [t,x2(t),color=red]], 0..50);
```



Due to model errors and noises, we need to stabilize the system around the reference trajectory ( $x1ref : x2ref : u1ref : u2ref$ ). In fact, we have  $Dx = Fx + Gu$  and  $Dxref = Fxref + Guref$ . Therefore, if we introduce the error  $e = x - xref$ , then  $e$  satisfies the following system  $De = Fe + G(u - uref)$ . If we choose the feedback  $u = uref + Ke$ , then the dynamic of the error satisfies  $De = (F + GK)e$ . Then, using the fact that the system is controllable, we can choose  $K$  so that the limit of  $e(t)$  tends to 0.

Finally, the closed-loop system is defined by  $Dx = Fx + G(uref + Ke)$ , i.e.,

$$Dx = (F + GK)x + Guref - GKxref.$$

For instance, we choose the closed-loop gain  $K$  as follows:

```
> K := evalm([[ -0.7425, 7.92 ], [ -0.2475, 0 ]]);
```

$$K := \begin{bmatrix} -0.7425 & 7.92 \\ -0.2475 & 0 \end{bmatrix}$$

Then,  $Fclosed = F + GK$  is defined by

```
> Fclosed := evalm(subs(c1=1,c2=2,c0=1.25,V0=1,theta=50,evalm(F+G&*K)));
```


$$F_{closed} := \begin{bmatrix} -1.000000000 & 7.92 \\ 0. & -2.000000000 \end{bmatrix}$$

and the characteristic polynomial is given by:

```
> Polcharact := collect(linalg[charpoly](Fclosed,lambda), lambda);
      Polcharact := 2.000000000 + lambda^2 + 3.000000000 lambda
> factor(Polcharact);
      (lambda + 1.000000000)(lambda + 2.000000000)
```

Thus, the eigenvalues of  $F_{closed}$  are  $-1$  and  $-2$ .

```
> Ge := subs([c1=1,c2=2,c0=1.25,V0=1,theta=50], evalm(G));
      Ge := [ 1 1
             -0.25 0.75 ]
> GKe := subs([c1=1,c2=2,c0=1.25,V0=1,theta=50], Mult(G, K, Alg));
      GKe := [ -0.9900000000  7.920000000
                0.            -1.980000000 ]
> ODE2 := {seq(evalm(ApplyMatrix(evalm(D-Fclosed), [x1(t),x2(t)]),
> Alg)-ApplyMatrix(Ge, [u1ref,u2ref],Alg)+ApplyMatrix(GKe,
> [x1ref,x2ref], Alg))[i,1]=0,i=1..2)};
      ODE2 := {2.000000000 x2(t) + (d/dt x2(t)) - 0.3714285714 e^(-t/7) = 0, 1.000000000 x1(t)
      + (d/dt x1(t)) - 7.92 x2(t) - 0.09000000000 e^(-t/10) + 1.584000000 e^(-t/7) = 0}
> dsol2 := dsolve(ODE2 union IC, {x1(t),x2(t)}, type=numeric, stiff=true):
> plots[odeplot](dsol2, [[t,x1(t), color=blue], [t,x2(t),color=red]], 0..50);
```



Now, let us consider the case where there are some perturbations in the system represented by the terms  $\exp(-10t)\sin(10t)$  in the right hand side of the equations of the system:

```
> ODE3 := {seq(ApplyMatrix(subs([c1=1,c2=2,c0=1.25,V0=1,theta=50], evalm(R)),
> [x1(t),x2(t),u1ref,u2ref], Alg)[i,1]=exp(-10*t)*sin(10*t)^2,i=1..2)};
```


$$ODE3 := \left\{ \frac{1}{50} x_2(t) + \left( \frac{d}{dt} x_2(t) \right) + 0.02457142857 e^{(-\frac{t}{7})} = e^{(-10t)} \sin(10t)^2, \right.$$

$$\left. \frac{1}{100} x_1(t) + \left( \frac{d}{dt} x_1(t) \right) + 0.009000000000 e^{(-\frac{t}{10})} = e^{(-10t)} \sin(10t)^2 \right\}$$

```
> dsol3 := dsolve(ODE3 union IC, {x1(t),x2(t)}, type=numeric, stiff=true):
```


Then, the open-loop  $x_1$  and  $x_2$  are the following

```
> plots[odeplot](dsol3, [[t,x1(t), color=blue], [t,x2(t),color=red]], 0..50);
```



whereas the following closed-loop  $x_1$  and  $x_2$  are:

```
> ODE4 := {seq(evalm(ApplyMatrix(evalm(D-Fclosed), [x1(t),x2(t)], Alg)-
> ApplyMatrix(Ge, [u1ref,u2ref], Alg)+ApplyMatrix(GKe, [x1ref,x2ref],
> Alg))[i,1]=exp(-10*t)*sin(10*t)^2,i=1..2)}:
> dsol4 := dsolve(ODE4 union IC, {x1(t),x2(t)}, type=numeric, stiff=true):
> plots[odeplot](dsol4, [[t,x1(t), color=blue], [t,x2(t),color=red]], 0..50);
```



## 1.5 Optimal control problems

Let us consider some linear quadratic problems for the stirred tank. (The definition of the cost function follows H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, p. 227.) First of all, let us define the following matrix  $Q$ :

```
> Q := evalm([[sigma1,0],[0,sigma2]]);
```

$$Q := \begin{bmatrix} \sigma_1 & 0 \\ 0 & \sigma_2 \end{bmatrix}$$

( $\sigma_1, \sigma_2$  are positive real numbers) which corresponds to the cost  $y^T Q y$ , where  $y$  is the output of the system defined in Section *Observability, input-output behaviour*. Using the fact that  $y = R y x$ , we then obtain the cost  $x^T C_1 x$  on the state  $x$  of the system, where the matrix  $C_1$  is defined by:

```
> C1 := evalm(linalg[transpose](Ry) &* Q &* Ry);
```

$$C_1 := \begin{bmatrix} \frac{\sigma_1}{4\theta^2} & 0 \\ 0 & \sigma_2 \end{bmatrix}$$

We also add to the previous cost an additional cost on the input  $u$  defined by  $u^T C_2 u$ , where  $C_2$  is the following matrix ( $\rho, \rho_1, \rho_2$  are positive real numbers):

```
> C2 := evalm([[rho*rho1,0],[0,rho*rho2]]);
```

$$C_2 := \begin{bmatrix} \rho \rho_1 & 0 \\ 0 & \rho \rho_2 \end{bmatrix}$$

Hence, we finally obtain the quadratic cost  $I = (x_1 : x_2 : u_1 : u_2) C (x_1 : x_2 : u_1 : u_2)^T$ , where  $C$  is defined by:

```
> C := linalg[diag](C1, C2);
```

$$C := \begin{bmatrix} \frac{\sigma 1}{4\theta^2} & 0 & 0 & 0 \\ 0 & \sigma 2 & 0 & 0 \\ 0 & 0 & \rho \rho 1 & 0 \\ 0 & 0 & 0 & \rho \rho 2 \end{bmatrix}$$

Now, let us consider the problem of minimizing the Lagrangian defined by  $\frac{1}{2}$  of the integral of  $I$  between 0 and  $T$  under the differential constraint formed by the system equations. Moreover, we consider the initial conditions of the system  $x1(0) = x10$  and  $x2(0) = x20$ . We shall see later that we can also add a cost on the terminal state  $x(T)$ .

Using the fact that the system is generically parametrizable (except the particular case  $c0 = c1 = c2$ ), we can substitute the parametrization *Ext*[3] of the system into the Lagrangian and into the terminal cost in order to obtain a purely variational problem without differential constraint on the free parameters  $\xi_1$  and  $\xi_2$  of the parametrization (see Section *Controllability, parametrizability, flatness, Brunovský canonical form*).

Therefore, the conditions of optimality can be obtained by computing the Euler-Lagrange equations for the new Lagrangian. For more details on the theory, we refer the reader to A. Quadrat, *Analyse algebrique des systemes lineaires multidimensionnels*, PhD Thesis, Ecole Nationale des Ponts et Chaussees, 1999, as well as J.-F. Pommaret, A. Quadrat, *A differential operator approach to multidimensional optimal control*, to appear in International Journal of Control, 2004.

In order to do that, we first add the constants  $\sigma 1$ ,  $\sigma 2$ ,  $\rho$ ,  $\rho 1$ , and  $\rho 2$  to the ones of the Ore algebra *Alg*.

```
> Alg := DefineOreAlgebra(diff=[D,t], polynom=[t],
> comm=[theta,V0,c0,c1,c2,sigma1,sigma2,rho,rho1,rho2]):
```

Then, we compute the corresponding Euler-Lagrange equations by using the command *LQEquations*.

```
> evalm(R);
```

$$\begin{bmatrix} D + \frac{1}{2\theta} & 0 & -1 & -1 \\ 0 & D + \frac{1}{\theta} & -\frac{c1 - c0}{V0} & -\frac{c2 - c0}{V0} \end{bmatrix}$$

```
> LQP := LQEquations(R, C, Alg):
```

The result *LQP* is a list with three entries (the third of which is the parametrization of the system which we are going to use at the end of this discussion). The first entry of *LQP* is a matrix that contains the two Euler-Lagrange equations for  $\xi_1$  and  $\xi_2$ , namely:

```
> LQP[1][1,1];
```

$$\begin{aligned}
& \xi_1(t) V0^2 \sigma1 c2^2 + \xi_2(t) V0^2 \sigma1 c1^2 + \xi_2(t) V0^2 \sigma1 c2^2 + \xi_1(t) V0^2 \sigma1 c1^2 \\
& - 2 \xi_2(t) V0^2 \sigma1 c1 c2 + 4 \xi_2(t) \theta^2 \sigma2 c1^3 c2 - 4 \xi_2(t) \theta^2 \sigma2 c1^3 c0 \\
& + 4 \xi_2(t) \theta^2 \sigma2 c1^2 c2 c0 - 8 \xi_2(t) \theta^2 \sigma2 c1^2 c2^2 + 4 \xi_2(t) \theta^2 \sigma2 c1 c2^2 c0 \\
& + 4 \xi_2(t) \theta^2 \sigma2 c1^2 c0^2 - 8 \xi_2(t) \theta^2 \sigma2 c1 c0^2 c2 + 4 \xi_2(t) \theta^2 \sigma2 c2^2 c0^2 \\
& - 2 \xi_2(t) V0^2 \rho \rho1 c1 c0 + 2 \xi_2(t) V0^2 \rho \rho1 c1 c2 - \xi_2(t) V0^2 \rho \rho1 c2^2 \\
& - \xi_2(t) V0^2 \rho \rho2 c1^2 + \xi_2(t) V0^2 \rho \rho1 c0^2 + \xi_2(t) V0^2 \rho \rho2 c0^2 \\
& - 8 \xi_1(t) \theta^2 \sigma2 c1^2 c2 c0 + 16 \xi_1(t) \theta^2 \sigma2 c1 c2^2 c0 - 8 \xi_1(t) \theta^2 \sigma2 c1 c0^2 c2 \\
& + \xi_1(t) V0^2 \rho \rho1 c2^2 + \xi_1(t) V0^2 \rho \rho1 c0^2 + \xi_1(t) V0^2 \rho \rho2 c1^2 + \xi_1(t) V0^2 \rho \rho2 c0^2 \\
& - 2 \xi_1(t) V0^2 \sigma1 c1 c2 + 4 \xi_1(t) \theta^2 \sigma2 c1^2 c2^2 - 8 \xi_1(t) \theta^2 \sigma2 c1 c2^3 \\
& + 4 \xi_1(t) \theta^2 \sigma2 c1^2 c0^2 - 8 \xi_1(t) \theta^2 \sigma2 c2^3 c0 + 4 \xi_1(t) \theta^2 \sigma2 c2^2 c0^2 \\
& - 4 \xi_1(t) V0^2 \rho \rho2 c2 c0 - 4 \xi_1(t) V0^2 \rho \rho2 c1 c2 + 4 \xi_1(t) \theta^2 \sigma2 c2^4 \\
& - 2 \xi_1(t) V0^2 \rho \rho1 c2 c0 + 2 \xi_1(t) V0^2 \rho \rho2 c1 c0 + 4 \xi_1(t) V0^2 \rho \rho2 c2^2 \\
& + 4 \xi_2(t) \theta^2 \sigma2 c1 c2^3 - 4 \xi_2(t) \theta^2 \sigma2 c2^3 c0 - 2 \xi_2(t) V0^2 \rho \rho2 c2 c0 \\
& + 2 \xi_2(t) V0^2 \rho \rho2 c1 c2 + 2 \left( \frac{d}{dt} \xi_2(t) \right) V0^2 \rho \rho1 c1 \theta c2 \\
& - 2 \left( \frac{d}{dt} \xi_2(t) \right) V0^2 \rho \rho1 c1 \theta c0 - 2 \left( \frac{d}{dt} \xi_2(t) \right) V0^2 \rho \rho1 \theta c2^2 \\
& + 2 \left( \frac{d}{dt} \xi_2(t) \right) V0^2 \rho \rho1 \theta c2 c0 + 2 \left( \frac{d}{dt} \xi_2(t) \right) V0^2 \rho \rho2 c1^2 \theta \\
& - 2 \left( \frac{d}{dt} \xi_2(t) \right) V0^2 \rho \rho2 c1 \theta c2 - 2 \left( \frac{d}{dt} \xi_2(t) \right) V0^2 \rho \rho2 c1 \theta c0 \\
& + 2 \left( \frac{d}{dt} \xi_2(t) \right) V0^2 \rho \rho2 \theta c2 c0 + 8 \left( \frac{d^2}{dt^2} \xi_1(t) \right) V0^2 \rho \rho2 c1 \theta^2 c2 \\
& - 4 \left( \frac{d^2}{dt^2} \xi_1(t) \right) V0^2 \rho \rho2 \theta^2 c2^2 - 4 \left( \frac{d^2}{dt^2} \xi_1(t) \right) V0^2 \rho \rho2 c1^2 \theta^2
\end{aligned}$$

> LQP[1][2,1];

$$\begin{aligned}
& -2 \left( \frac{d}{dt} \xi_1(t) \right) V0^2 \rho \rho2 \theta c2 c0 + \xi_1(t) V0^2 \sigma1 c2^2 + \xi_2(t) V0^2 \sigma1 c1^2 + \xi_2(t) V0^2 \sigma1 c2^2 \\
& + 4 \xi_2(t) \theta^2 \sigma2 c1^4 + \xi_1(t) V0^2 \sigma1 c1^2 - 2 \xi_2(t) V0^2 \sigma1 c1 c2 - 8 \xi_2(t) \theta^2 \sigma2 c1^3 c2 \\
& - 8 \xi_2(t) \theta^2 \sigma2 c1^3 c0 + 16 \xi_2(t) \theta^2 \sigma2 c1^2 c2 c0 + 4 \xi_2(t) \theta^2 \sigma2 c1^2 c2^2 \\
& - 8 \xi_2(t) \theta^2 \sigma2 c1 c2^2 c0 + 4 \xi_2(t) \theta^2 \sigma2 c1^2 c0^2 - 8 \xi_2(t) \theta^2 \sigma2 c1 c0^2 c2 \\
& + 4 \xi_2(t) \theta^2 \sigma2 c2^2 c0^2 + 2 \xi_2(t) V0^2 \rho \rho1 c2 c0 - 4 \xi_2(t) V0^2 \rho \rho1 c1 c0 \\
& - 4 \xi_2(t) V0^2 \rho \rho1 c1 c2 + 4 \xi_2(t) V0^2 \rho \rho1 c1^2 + \xi_2(t) V0^2 \rho \rho1 c2^2 \\
& + \xi_2(t) V0^2 \rho \rho2 c1^2 + \xi_2(t) V0^2 \rho \rho1 c0^2 + \xi_2(t) V0^2 \rho \rho2 c0^2 \\
& - 2 \xi_2(t) V0^2 \rho \rho2 c1 c0 + 4 \xi_1(t) \theta^2 \sigma2 c1^2 c2 c0 + 4 \xi_1(t) \theta^2 \sigma2 c1 c2^2 c0 \\
& - 8 \xi_1(t) \theta^2 \sigma2 c1 c0^2 c2 - \xi_1(t) V0^2 \rho \rho1 c2^2 + \xi_1(t) V0^2 \rho \rho1 c0^2 \\
& - \xi_1(t) V0^2 \rho \rho2 c1^2 + \xi_1(t) V0^2 \rho \rho2 c0^2 - 2 \xi_1(t) V0^2 \sigma1 c1 c2 \\
& + 4 \xi_1(t) \theta^2 \sigma2 c1^3 c2 - 4 \xi_1(t) \theta^2 \sigma2 c1^3 c0 - 8 \xi_1(t) \theta^2 \sigma2 c1^2 c2^2 \\
& + 4 \xi_1(t) \theta^2 \sigma2 c1 c2^3 + 4 \xi_1(t) \theta^2 \sigma2 c1^2 c0^2 - 4 \xi_1(t) \theta^2 \sigma2 c2^3 c0 \\
& + 4 \xi_1(t) \theta^2 \sigma2 c2^2 c0^2 - 2 \xi_1(t) V0^2 \rho \rho1 c1 c0 - 2 \xi_1(t) V0^2 \rho \rho2 c2 c0 \\
& + 2 \xi_1(t) V0^2 \rho \rho1 c1 c2 + 2 \xi_1(t) V0^2 \rho \rho2 c1 c2 - 2 \left( \frac{d}{dt} \xi_1(t) \right) V0^2 \rho \rho1 c1 \theta c2 \\
& + 2 \left( \frac{d}{dt} \xi_1(t) \right) V0^2 \rho \rho1 c1 \theta c0 + 2 \left( \frac{d}{dt} \xi_1(t) \right) V0^2 \rho \rho1 \theta c2^2 \\
& - 2 \left( \frac{d}{dt} \xi_1(t) \right) V0^2 \rho \rho1 \theta c2 c0 - 2 \left( \frac{d}{dt} \xi_1(t) \right) V0^2 \rho \rho2 c1^2 \theta \\
& + 2 \left( \frac{d}{dt} \xi_1(t) \right) V0^2 \rho \rho2 c1 \theta c2 + 2 \left( \frac{d}{dt} \xi_1(t) \right) V0^2 \rho \rho2 c1 \theta c0 \\
& + 8 \left( \frac{d^2}{dt^2} \xi_2(t) \right) V0^2 \rho \rho1 c1 \theta^2 c2 - 4 \left( \frac{d^2}{dt^2} \xi_2(t) \right) V0^2 \rho \rho1 \theta^2 c2^2 \\
& - 4 \left( \frac{d^2}{dt^2} \xi_2(t) \right) V0^2 \rho \rho1 c1^2 \theta^2
\end{aligned}$$

Hence, a necessary condition for optimality is defined by  $LQP[1][1,1] = 0$  and  $LQP[1][2,1] = 0$ . Moreover,



the boundary terms coming from the integrations by parts are defined by the second matrix  $LQP[2]$ :

$$\begin{aligned}
&> \text{Boundary} := \text{collect}(LQP[2], \{\text{delta}[\text{xi}[1]](t), \text{delta}[\text{xi}[2]](t)\}); \\
&\text{Boundary} := 2\theta V0^2 \rho(-c2 \rho2 \xi_2(t) c0 - 3 c1 \rho2 \xi_1(t) c2 + c1 \rho2 \xi_1(t) c0 + c1^2 \rho2 \xi_1(t) \\
&+ 2 c1^2 \rho2 (\frac{d}{dt} \xi_1(t)) \theta - 4 c1 \rho2 (\frac{d}{dt} \xi_1(t)) c2 \theta - c1^2 \rho2 \xi_2(t) + c1 \rho2 \xi_2(t) c0 \\
&+ 2 c2^2 \rho2 \xi_1(t) - c2 \rho2 \xi_1(t) c0 + 2 c2^2 \rho2 (\frac{d}{dt} \xi_1(t)) \theta + c2 \rho2 \xi_2(t) c1) \delta_{\xi_1}(t) + 2 \\
&\theta V0^2 \rho(-c1 \rho1 \xi_1(t) c0 - c1 \rho1 \xi_2(t) c0 + 2 c1^2 \rho1 \xi_2(t) - 3 c1 \rho1 \xi_2(t) c2 \\
&+ 2 c1^2 \rho1 (\frac{d}{dt} \xi_2(t)) \theta - 4 c1 \rho1 (\frac{d}{dt} \xi_2(t)) c2 \theta - c2^2 \rho1 \xi_1(t) + c2 \rho1 \xi_1(t) c0 \\
&+ c2 \rho1 \xi_2(t) c0 + c2^2 \rho1 \xi_2(t) + 2 c2^2 \rho1 (\frac{d}{dt} \xi_2(t)) \theta + c1 \rho1 \xi_1(t) c2) \delta_{\xi_2}(t)
\end{aligned}$$

where  $\delta_{\xi_1}(t)$  and  $\delta_{\xi_2}(t)$  are the variational of  $\xi_1(t)$  and  $\xi_2(t)$  resp.. In order to obtain the initial and final conditions of the system  $LQP[1][1,1] = 0$ ,  $LQP[1][2,1] = 0$ , we need to consider whether or not the terminal value of  $x(T)$  is fixed or if there exists a terminal cost  $x(T)^T C x(T)$  on  $x(T)$ .

1. In the first case, we do not need to consider the boundary terms *Boundary* previously computed. However, we need to transfer the final conditions on  $x(T)$  to final conditions on  $\xi_1$  and  $\xi_2$  and their derivatives by using the parametrization *Ext[3]* of the system.
2. In the second case, we need to substitute the parametrization of the system into the terminal cost  $x(T)^T C x(T)$  and compute its variation. We obtain some linear terms in the variations  $\delta_{\xi_1}(t)$  and  $\delta_{\xi_2}(t)$  that we need to add to the boundary terms that we have already computed in *Boundary*.

For instance, let us consider the case where there is no cost on the terminal state  $x(T)$  and  $x(T)$  is free. Then, the first final condition is defined by:

$$\begin{aligned}
&> \text{B1} := \text{subs}(t=T, \text{subs}([\text{delta}[\text{xi}[1]](t)=1, \text{delta}[\text{xi}[2]](t)=0], \text{Boundary})); \\
&\text{B1} := 2\theta V0^2 \rho(-c2 \rho2 \xi_2(T) c0 - 3 c1 \rho2 \xi_1(T) c2 + c1 \rho2 \xi_1(T) c0 + c1^2 \rho2 \xi_1(T) \\
&+ 2 c1^2 \rho2 (\frac{d}{dT} \xi_1(T)) \theta - 4 c1 \rho2 (\frac{d}{dT} \xi_1(T)) c2 \theta - c1^2 \rho2 \xi_2(T) + c1 \rho2 \xi_2(T) c0 \\
&+ 2 c2^2 \rho2 \xi_1(T) - c2 \rho2 \xi_1(T) c0 + 2 c2^2 \rho2 (\frac{d}{dT} \xi_1(T)) \theta + c2 \rho2 \xi_2(T) c1)
\end{aligned}$$

The second final condition is defined by:

$$\begin{aligned}
&> \text{B2} := \text{subs}(t=T, \text{subs}([\text{delta}[\text{xi}[1]](t)=0, \text{delta}[\text{xi}[2]](t)=1], \text{Boundary})); \\
&\text{B2} := 2\theta V0^2 \rho(-c1 \rho1 \xi_1(T) c0 - c1 \rho1 \xi_2(T) c0 + 2 c1^2 \rho1 \xi_2(T) - 3 c1 \rho1 \xi_2(T) c2 \\
&+ 2 c1^2 \rho1 (\frac{d}{dT} \xi_2(T)) \theta - 4 c1 \rho1 (\frac{d}{dT} \xi_2(T)) c2 \theta - c2^2 \rho1 \xi_1(T) + c2 \rho1 \xi_1(T) c0 \\
&+ c2 \rho1 \xi_2(T) c0 + c2^2 \rho1 \xi_2(T) + 2 c2^2 \rho1 (\frac{d}{dT} \xi_2(T)) \theta + c1 \rho1 \xi_1(T) c2)
\end{aligned}$$

The initial conditions are obtained by transferring the initial conditions  $x1(0) = x10$ ,  $x2(0) = x20$ ,  $D(x1)(0) = x1dot0$ , and  $D(x2)(0) = x2dot0$  to initial conditions on  $\xi_1$  and  $\xi_2$  and their derivatives. We obtain:

$$\begin{aligned}
&> \text{IC} := \{\text{subs}(t=0, \text{Parametrization}(R, \text{Alg})[1,1]=x10), \text{subs}(t=0, \\
&+ \text{Parametrization}(R, \text{Alg})[2,1]=x20)\}; \\
&\text{IC} := \{-2 \xi_1(0) c1 \theta V0 + 2 \xi_1(0) c2 \theta V0 - 2 \xi_2(0) c1 \theta V0 + 2 \xi_2(0) c2 \theta V0 = x10, \\
&-2 \xi_1(0) c2 c1 \theta + 2 \xi_1(0) c2^2 \theta + 2 \xi_1(0) c0 c1 \theta - 2 \xi_1(0) c0 c2 \theta - 2 \xi_2(0) c1^2 \theta \\
&+ 2 \xi_2(0) c2 c1 \theta + 2 \xi_2(0) c0 c1 \theta - 2 \xi_2(0) c0 c2 \theta = x20\}
\end{aligned}$$

In order to obtain the optimal trajectory, we then need to integrate the following system of ordinary differential equations

```
> ODEs := {LQP[1][1,1],LQP[1][2,1]}:
```

with the following initial and final conditions:

```
> BoundaryConditions := {B1,B2,op(IC)}:
```

Such a system is in general difficult to integrate via symbolic integrations (see however some simple examples in the *OreModules* library). Hence, we need to know the particular values of the constants  $\theta$ ,  $V0$ ,  $c0$ ,  $c1$ ,  $c2$ ,  $\sigma1$ ,  $\sigma2$ ,  $\rho$ ,  $\rho1$ ,  $\rho2$  of the system, the final time  $T$  and to integrate it numerically.

The values of the constants are chosen as above:

```
> const_values := [c0=1.25,c1=1,c2=2,V0=1,theta=50];
const_values := [c0 = 1.25, c1 = 1, c2 = 2, V0 = 1, theta = 50]
```

We substitute these values, appropriate values for the cost function, the initial conditions and the final time into *ODEs* and *BoundaryConditions* given by *LQEquations*:

```
> BVP := subs([op(const_values),sigma1=50,sigma2=0.02,rho=0.1,rho1=1/3,rho2=3,
> x10=0.1,x20=0,T=50], ODEs union map(convert, BoundaryConditions, D));
```

$$\begin{aligned} BVP := & \{100 \xi_1(0) + 100 \xi_2(0) = 0.1, \\ & -2.500000000 \xi_1(50) + 4.166666668 \xi_2(50) + 333.3333334 D(\xi_2)(50), \\ & 75.00 \xi_1(0) - 25.00 \xi_2(0) = 0, -5.0000000 (\frac{d}{dt} \xi_1(t)) + 12.337500 \xi_1(t) \\ & + 62.570832 \xi_2(t) - 333.3333333 (\frac{d^2}{dt^2} \xi_2(t)), \\ & -7.500 \xi_2(50) + 52.500 \xi_1(50) + 3000.0 D(\xi_1)(50), \\ & -3000.0 (\frac{d^2}{dt^2} \xi_1(t)) + 163.4374993 \xi_1(t) + 12.337500 \xi_2(t) + 5.00000000 (\frac{d}{dt} \xi_2(t)) \} \end{aligned}$$

This defines the boundary value problem to be solved numerically:

```
> dsol5 := dsolve(BVP, {xi[1](t),xi[2](t)}, type=numeric);
dsol5 := proc(x_bvp) ... end proc
```


In order to find the corresponding trajectories  $x1(t)$ ,  $x2(t)$ ,  $u1(t)$ ,  $u2(t)$  of the system, we have to adjust the parametrization *LQP[3]* of the system to the chosen values for the constants:

```
> P_ex := subs([op(const_values),xi=eta], LQP[3]);
```

$$P\_ex := \begin{bmatrix} 100 \eta_1(t) + 100 \eta_2(t) \\ 75.00 \eta_1(t) - 25.00 \eta_2(t) \\ -0.75 \eta_1(t) + 1.25 \eta_2(t) + 100 (\frac{d}{dt} \eta_2(t)) \\ 1.75 \eta_1(t) + 100 (\frac{d}{dt} \eta_1(t)) - 0.25 \eta_2(t) \end{bmatrix}$$


We have also renamed the free parameters  $\xi_1$ ,  $\xi_2$  of the parametrization into  $\eta_1$ ,  $\eta_2$  because we want to plug in the solutions  $\xi_1(t)$ ,  $\xi_2(t)$  of the boundary value problem. Then, the first two components of the parametrization give  $x1(t)$ ,  $x2(t)$ :

```
> plots[odeplot](dsol5, [[t,subs([eta[1](t)=xi[1](t), eta[2](t)=xi[2](t)],
> P_ex[1,1]), color=blue], [t,subs([eta[1](t)=xi[1](t), eta[2](t)=xi[2](t)],
> P_ex[2,1]), color=red]], 0..50);
```



The third and the fourth component of the parametrization yield the trajectories for  $u_1(t)$  and  $u_2(t)$ :

```
> plots[odeplot](dsol5, [[t,subs([eta[1](t)=xi[1](t), eta[2](t)=xi[2](t)],
> P_ex[3,1]), color=blue], [t,subs([eta[1](t)=xi[1](t), eta[2](t)=xi[2](t)],
> P_ex[4,1]), color=red]], 0..50);
```



We have chosen  $\rho = 0.1$  here. The plots for  $x_1(t)$ ,  $x_2(t)$  resp.  $u_1(t)$ ,  $u_2(t)$  above can be compared to the ones given in H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, on pages 228 resp. 229.

## 2 Stirred tank: a discrete-time model

We now consider the discrete-time description of the stirred tank given in H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, p. 449. In order to do that, we first introduce the algebra  $Alg2$  of polynomials in the shift operator  $\sigma$ , where  $\sigma x(i) = x(i + 1)$ .

```
> Alg2 := DefineOreAlgebra(shift=[sigma,i], polynom=[i],
> comm=[theta,V0,c0,c1,c2,kappa]):
```

Then, we introduce the system matrix:

```
> R2 := evalm([[sigma-kappa,0,-2*theta*(1-kappa),-2*theta*(1-kappa)],
> [0,sigma-kappa^2,-theta*(c1-c0)*(1-kappa^2)/V0,-theta*(c2-c0)*(1-kappa^2)/V0]]);
```

$$R2 := \begin{bmatrix} \sigma - \kappa & 0 & -2\theta(1 - \kappa) & -2\theta(1 - \kappa) \\ 0 & \sigma - \kappa^2 & -\frac{\theta(c1 - c0)(1 - \kappa^2)}{V0} & -\frac{\theta(c2 - c0)(1 - \kappa^2)}{V0} \end{bmatrix}$$

Let us check whether or not this new system is parametrizable and flat.

```
> R2_adj := Involution(R2, Alg2);
```

$$R2\_adj := \begin{bmatrix} -\sigma - \kappa & 0 \\ 0 & -\sigma - \kappa^2 \\ -2\theta + 2\theta\kappa & -\frac{\theta c1}{V0} + \frac{\theta c1 \kappa^2}{V0} + \frac{\theta c0}{V0} - \frac{\theta c0 \kappa^2}{V0} \\ -2\theta + 2\theta\kappa & -\frac{\theta c2}{V0} + \frac{\theta c2 \kappa^2}{V0} + \frac{\theta c0}{V0} - \frac{\theta c0 \kappa^2}{V0} \end{bmatrix}$$

```
> Ext2 := Exti(R2_adj, Alg2, 1);
```

$$Ext2 := \left[ \begin{array}{c} \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} \sigma - \kappa, 0, -2\theta + 2\theta\kappa, -2\theta + 2\theta\kappa \\ 0, V0\sigma - \kappa^2 V0, \kappa^2 c1\theta - c1\theta - c0\theta\kappa^2 + \theta c0, -c2\theta - c0\theta\kappa^2 + \theta c0 + \kappa^2 c2\theta \end{bmatrix} \\ , \\ \begin{bmatrix} -2c1\theta V0 + 2c2\theta V0 - 2c2\theta V0\kappa + 2c1\theta V0\kappa, \\ -2c1\theta V0 + 2c2\theta V0 - 2c2\theta V0\kappa + 2c1\theta V0\kappa \\ \left[ c0\theta c2\kappa^2 - c0\theta c1\kappa^2 + c2^2\theta - \theta c2^2\kappa^2 - c2c1\theta - c0c2\theta + c0c1\theta \right. \\ \left. + c2\theta c1\kappa^2, c0\theta c2\kappa^2 - c0\theta c1\kappa^2 + \theta c1^2\kappa^2 + c2c1\theta - c0c2\theta + c0c1\theta \right. \\ \left. - c2\theta c1\kappa^2 - c1^2\theta \right] \\ \left[ c0\kappa V0 + c2\kappa^2 V0 - c0\kappa^2 V0 - c2\kappa V0, \right. \\ \left. -\sigma c1 V0 + c2\sigma V0 + c0\kappa V0 + c1\kappa^2 V0 - c0\kappa^2 V0 - c2\kappa V0 \right] \\ \left[ c1\kappa V0 - c0\kappa V0 - \sigma c1 V0 + c2\sigma V0 - c2\kappa^2 V0 + c0\kappa^2 V0, \right. \\ \left. c1\kappa V0 - c0\kappa V0 - c1\kappa^2 V0 + c0\kappa^2 V0 \right] \end{array} \right]$$

The first matrix  $Ext2[1]$  of  $Ext2$  is the identity matrix, and thus, the  $Alg2$ -module  $M2$  associated with the system is torsion-free. Therefore, the system is generically controllable. Moreover,  $Ext2[3]$  is a parametrization of the system. Equivalently, we have:

```
> Parametrization(R2, Alg2);
```

$$\begin{aligned}
& \left[ -2\xi_1(i) c1 \theta V0 + 2\xi_1(i) c2 \theta V0 - 2\xi_1(i) c2 \theta V0 \kappa + 2\xi_1(i) c1 \theta V0 \kappa \right. \\
& \quad \left. - 2\xi_2(i) c1 \theta V0 + 2\xi_2(i) c2 \theta V0 - 2\xi_2(i) c2 \theta V0 \kappa + 2\xi_2(i) c1 \theta V0 \kappa \right] \\
& \left[ \xi_1(i) c0 \theta c2 \kappa^2 - \xi_1(i) c0 \theta c1 \kappa^2 + \xi_1(i) c2^2 \theta - \xi_1(i) \theta c2^2 \kappa^2 - \xi_1(i) c2 c1 \theta \right. \\
& \quad - \xi_1(i) c0 c2 \theta + \xi_1(i) c0 c1 \theta + \xi_1(i) c2 \theta c1 \kappa^2 + \xi_2(i) c0 \theta c2 \kappa^2 \\
& \quad - \xi_2(i) c0 \theta c1 \kappa^2 + \xi_2(i) \theta c1^2 \kappa^2 + \xi_2(i) c2 c1 \theta - \xi_2(i) c0 c2 \theta + \xi_2(i) c0 c1 \theta \\
& \quad \left. - \xi_2(i) c2 \theta c1 \kappa^2 - \xi_2(i) c1^2 \theta \right] \\
& \left[ \xi_1(i) c0 \kappa V0 + \xi_1(i) c2 \kappa^2 V0 - \xi_1(i) c0 \kappa^2 V0 - \xi_1(i) c2 \kappa V0 + \xi_2(i) c0 \kappa V0 \right. \\
& \quad - \xi_2(i) c2 \kappa V0 + \xi_2(i) c1 \kappa^2 V0 - \xi_2(i) c0 \kappa^2 V0 - \xi_2(i+1) c1 V0 \\
& \quad \left. + \xi_2(i+1) c2 V0 \right] \\
& \left[ \xi_1(i) c1 \kappa V0 - \xi_1(i) c0 \kappa V0 - \xi_1(i) c2 \kappa^2 V0 + \xi_1(i) c0 \kappa^2 V0 - \xi_1(i+1) c1 V0 \right. \\
& \quad \left. + \xi_1(i+1) c2 V0 + \xi_2(i) c1 \kappa V0 - \xi_2(i) c0 \kappa V0 - \xi_2(i) c1 \kappa^2 V0 + \xi_2(i) c0 \kappa^2 V0 \right]
\end{aligned}$$

Moreover, the  $Alg2$ -module  $M2$  associated with the system is projective because  $M2$  is torsion-free and every torsion-free module over  $Alg2$  is projective as  $Alg2$  is a hereditary ring. This result can be directly verified by checking the existence of a right-inverse of the matrix  $R2$ .

> T2 := map(factor, RightInverse(R2, Alg2));

$$T2 := \begin{bmatrix} 0 & 0 \\ 0 & V0 \\ \frac{-c2 + c0}{2\theta(c1 - c2)(-1 + \kappa)} & \frac{V0}{\theta(-1 + \kappa)(\kappa + 1)(c1 - c2)} \\ \frac{-c1 + c0}{-2\theta(c1 - c2)(-1 + \kappa)} & \frac{V0}{-\theta(-1 + \kappa)(\kappa + 1)(c1 - c2)} \end{bmatrix}$$

> Mult(R2, T2, Alg2);

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

The fact that the system is time-invariant implies that the  $Alg2$ -module  $M2$  is free. Therefore, the system is generically flat. Let us compute a flat output of the system.

> S2 := map(factor, LeftInverse(Ext2[3], Alg2));

$$S2 := \begin{bmatrix} \frac{-c1 + c0}{2(c1 - c2)^2(-1 + \kappa)\theta V0} & -\frac{1}{\theta(c1 - c2)^2(-1 + \kappa)(\kappa + 1)} & 0 & 0 \\ \frac{-c2 + c0}{2(c1 - c2)^2(-1 + \kappa)\theta V0} & \frac{1}{\theta(c1 - c2)^2(-1 + \kappa)(\kappa + 1)} & 0 & 0 \end{bmatrix}$$

> Mult(S2, Ext2[3], Alg2);

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Hence, a flat output of the system is defined by  $(\xi1 : \xi2)^T = S2 (x1 : x2 : u1 : u2)^T$ , where  $\xi1$  and  $\xi2$  satisfy  $(x1 : x2 : u1 : u2)^T = Ext2[3] (\xi1 : \xi2)^T$ . More precisely, this flat output of the system is defined by:

> evalm([[xi1(i)], [xi2(i)]])=ApplyMatrix(S2, [x1(i), x2(i), u1(i), u2(i)], Alg2);

$$\begin{bmatrix} \xi_1(i) \\ \xi_2(i) \end{bmatrix} = \begin{bmatrix} -\frac{1}{2} \frac{(-c_1 + c_0) x_1(i)}{(c_1 - c_2)^2 (-1 + \kappa) \theta V_0} - \frac{x_2(i)}{\theta (c_1 - c_2)^2 (-1 + \kappa) (\kappa + 1)} \\ \frac{1}{2} \frac{(-c_2 + c_0) x_1(i)}{(c_1 - c_2)^2 (-1 + \kappa) \theta V_0} + \frac{x_2(i)}{\theta (c_1 - c_2)^2 (-1 + \kappa) (\kappa + 1)} \end{bmatrix}$$

By substituting  $(\xi_1 : \xi_2)^T = S_2 (x_1 : x_2 : u_1 : u_2)^T$  into  $(x_1 : x_2 : u_1 : u_2)^T = \text{Ext2}[3] (\xi_1 : \xi_2)^T$ , we obtain the following matrix:

> P2 := simplify(Mult(Ext2[3], S2, Alg2));

$$P_2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ \frac{-\kappa c_2 + c_0 \kappa + \sigma c_2 - \sigma c_0}{2(-1 + \kappa)(c_1 - c_2)\theta} & \frac{(\kappa^2 - \sigma) V_0}{(c_1 - c_2)\theta(-1 + \kappa^2)} & 0 & 0 \\ -\frac{\sigma c_1 - \kappa c_1 + c_0 \kappa - \sigma c_0}{2(-1 + \kappa)(c_1 - c_2)\theta} & -\frac{(\kappa^2 - \sigma) V_0}{(c_1 - c_2)\theta(-1 + \kappa^2)} & 0 & 0 \end{bmatrix}$$

Hence, we have obtained that  $(x_1 : x_2 : u_1 : u_2)^T = P_2 (x_1 : x_2 : u_1 : u_2)^T$ . From  $P_2$ , it is obvious that another flat output of the system is defined by  $\xi_1 = x_1$  and  $\xi_2 = x_2$ .

> evalm([[x1(i)], [x2(i)], [u1(i)], [u2(i)]]]=  
> ApplyMatrix(P2, [x1(i), x2(i), u1(i), u2(i)], Alg2);

$$\begin{bmatrix} x_1(i) \\ x_2(i) \\ u_1(i) \\ u_2(i) \end{bmatrix} = \begin{bmatrix} x_1(i) \\ x_2(i) \\ \left[ \frac{1}{2} \frac{(-\kappa c_2 + c_0 \kappa) x_1(i)}{(-1 + \kappa)(c_1 - c_2)\theta} + \frac{1}{2} \frac{(c_2 - c_0) x_1(i+1)}{(-1 + \kappa)(c_1 - c_2)\theta} + \frac{\kappa^2 V_0 x_2(i)}{(c_1 - c_2)\theta(-1 + \kappa^2)} - \frac{V_0 x_2(i+1)}{(c_1 - c_2)\theta(-1 + \kappa^2)} \right] \\ \left[ -\frac{1}{2} \frac{(-\kappa c_1 + c_0 \kappa) x_1(i)}{(-1 + \kappa)(c_1 - c_2)\theta} - \frac{1}{2} \frac{(c_1 - c_0) x_1(i+1)}{(-1 + \kappa)(c_1 - c_2)\theta} - \frac{\kappa^2 V_0 x_2(i)}{(c_1 - c_2)\theta(-1 + \kappa^2)} + \frac{V_0 x_2(i+1)}{(c_1 - c_2)\theta(-1 + \kappa^2)} \right] \end{bmatrix}$$

To conclude, we notice that the system is flat if  $c_1 \neq c_2$  and  $\kappa^2 \neq 1$ . However,  $\kappa = e^{(-\frac{\Delta}{2\theta})}$ , where  $\Delta$  is the constant length of the time interval of the sampler. In particular, if  $\Delta \neq 0$  and  $c_1 \neq c_2$ , then we obtain that the discrete-time model of the stirred tank is flat.

Then, as in the ordinary differential model, we can use the above parametrization in order to do motion planning and tracking.

Let us compute the Brunovsky canonical form for the system in the case  $c_1 \neq c_2$ ,  $\kappa^2 \neq 1$ .

> B2 := Brunovsky(R2, Alg2);

$$B2 := \begin{bmatrix} -\frac{-c1+c0}{2(c1-c2)^2(-1+\kappa)\theta V0}, -\frac{1}{\theta(c1-c2)^2(-1+\kappa)(\kappa+1)}, 0, 0 \\ \frac{(-c1+c0)\kappa}{2(c1-c2)^2(-1+\kappa)V0\theta}, -\frac{\kappa^2}{\theta(c1-c2)^2(-1+\kappa)(\kappa+1)}, 0, -\frac{1}{(c1-c2)V0} \\ \frac{-c2+c0}{2(c1-c2)^2(-1+\kappa)\theta V0}, \frac{1}{\theta(c1-c2)^2(-1+\kappa)(\kappa+1)}, 0, 0 \\ \frac{(-c2+c0)\kappa}{2(c1-c2)^2(-1+\kappa)V0\theta}, \frac{\kappa^2}{\theta(c1-c2)^2(-1+\kappa)(\kappa+1)}, -\frac{1}{(c1-c2)V0}, 0 \end{bmatrix}$$

We obtain the following transformation:

```
> evalm([[z1(i)], [v1(i)], [z2(i)], [v2(i)]])=
> ApplyMatrix(B2, [x1(i), x2(i), u1(i), u2(i)], Alg2);
```

$$\begin{bmatrix} z1(i) \\ v1(i) \\ z2(i) \\ v2(i) \end{bmatrix} = \begin{bmatrix} -\frac{1}{2} \frac{(-c1+c0)x1(i)}{(c1-c2)^2(-1+\kappa)\theta V0} - \frac{x2(i)}{\theta(c1-c2)^2(-1+\kappa)(\kappa+1)} \\ -\frac{1}{2} \frac{(-c1+c0)\kappa x1(i)}{(c1-c2)^2(-1+\kappa)V0\theta} - \frac{\kappa^2 x2(i)}{\theta(c1-c2)^2(-1+\kappa)(\kappa+1)} - \frac{u2(i)}{(c1-c2)V0} \\ \frac{1}{2} \frac{(-c2+c0)x1(i)}{(c1-c2)^2(-1+\kappa)\theta V0} + \frac{x2(i)}{\theta(c1-c2)^2(-1+\kappa)(\kappa+1)} \\ \frac{1}{2} \frac{(-c2+c0)\kappa x1(i)}{(c1-c2)^2(-1+\kappa)V0\theta} + \frac{\kappa^2 x2(i)}{\theta(c1-c2)^2(-1+\kappa)(\kappa+1)} - \frac{u1(i)}{(c1-c2)V0} \end{bmatrix}$$

Let us check that the new variables  $\zeta[i]$  satisfy the Brunovsky canonical form. In order to do that, modulo the system equations, we need to eliminate the variables  $x1$ ,  $x2$ ,  $u1$  and  $u2$  from the previous system.

```
> F2 := Elimination(linalg[stackmatrix](B2, R2), [x1,x2,u1,u2],
> [z1,v1,z2,v2,0,0], Alg2):
> ApplyMatrix(F2[1], [x1(i),x2(i),u1(i),u2(i)], Alg2)=
> ApplyMatrix(F2[2], [z1(i),v1(i),z2(i),v2(i)], Alg2);
```

$$\begin{bmatrix} 0 \\ 0 \\ u2(i) \\ u1(i) \\ x2(i) \\ x1(i) \end{bmatrix} = \begin{bmatrix} [-z2(i+1) + v2(i)] \\ [-z1(i+1) + v1(i)] \\ [(c0\kappa^2 V0 - c0\kappa V0 - c2\kappa^2 V0 + c1\kappa V0)z1(i) + (-c1 V0 + c2 V0)v1(i) \\ + (c1\kappa V0 - c0\kappa V0 - c1\kappa^2 V0 + c0\kappa^2 V0)z2(i)] \\ [(c0\kappa V0 + c2\kappa^2 V0 - c0\kappa^2 V0 - c2\kappa V0)z1(i) \\ + (-c0\kappa^2 V0 + c0\kappa V0 + c1\kappa^2 V0 - c2\kappa V0)z2(i) + (-c1 V0 + c2 V0)v2(i)] \\ [(c0\theta c2\kappa^2 - c0\theta c1\kappa^2 + c2^2\theta - \theta c2^2\kappa^2 - c2c1\theta - c0c2\theta + c0c1\theta \\ + c2\theta c1\kappa^2)z1(i) + (c0\theta c2\kappa^2 - c0\theta c1\kappa^2 + \theta c1^2\kappa^2 + c2c1\theta - c0c2\theta \\ + c0c1\theta - c2\theta c1\kappa^2 - c1^2\theta)z2(i)] \\ [(-2c1\theta V0 + 2c2\theta V0 - 2c2\theta V0\kappa + 2c1\theta V0\kappa)z1(i) \\ + (-2c1\theta V0 + 2c2\theta V0 - 2c2\theta V0\kappa + 2c1\theta V0\kappa)z2(i)] \end{bmatrix}$$

The first two equations show that the  $z_i$  and  $v_j$  satisfy a Brunovský canonical form.

Finally, let us study the case where  $c\theta = c1 = c2$ . Then, the system matrix becomes the following one:

```
> Sys2 := subs([c2=c0,c1=c0], evalm(R2));
```

$$Sys2 := \begin{bmatrix} \sigma - \kappa & 0 & -2\theta(1 - \kappa) & -2\theta(1 - \kappa) \\ 0 & \sigma - \kappa^2 & 0 & 0 \end{bmatrix}$$

Let us check whether or not the system is controllable, parametrizable and flat.

```
> Ext := Exti(Involution(Sys2, Alg2), Alg2, 1);
```

$$Ext := \left[ \begin{bmatrix} 1 & 0 \\ 0 & \sigma - \kappa^2 \end{bmatrix}, \begin{bmatrix} \sigma - \kappa & 0 & -2\theta + 2\theta\kappa & -2\theta + 2\theta\kappa \\ 0 & 1 & 0 & 0 \end{bmatrix}, \begin{bmatrix} -2\theta + 2\theta\kappa & 0 \\ 0 & 0 \\ 0 & 1 \\ -\sigma + \kappa & -1 \end{bmatrix} \right]$$

The first matrix  $Ext[1]$  is not the identity matrix, and thus, the system is not controllable, not parametrizable and not flat. The torsion elements in the  $Alg2$ -module associated with  $R2$  are defined by:

```
> TorsionElements(Sys2, [x1(i),x2(i),u1(i),u2(i)], Alg2);
```

$$[[ -\kappa^2 \theta_2(i) + \theta_2(i + 1) = 0 ], [ \theta_2(i) = x2(i) ]]$$

The controllable part of the system is defined by  $Ext[2]$  and it is parametrized by  $Ext[3]$ .

### 3 Stirred tank: a differential time-delay model

Now, we consider the extension of the stirred tank model by considering a transport time-delay  $\tau$  in the pipe. See H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, p. 450-451. We need to introduce the time-delay operator in the Ore algebra as follows:

```
> Alg3 := DefineOreAlgebra(diff=[D,t], dual_shift=[delta,s],
> polynom=[t,s], comm=[theta,V0,c0,c1,c2], shift_action=[delta,t,h]):
```

The new system is defined by the following matrix of operators:

```
> R3 := evalm([[D+1/(2*theta),0,-1,-1],
> [0,D+1/theta,-(c1-c0)*delta/V0,-(c2-c1)*delta/V0]]);
```

$$R3 := \begin{bmatrix} D + \frac{1}{2\theta} & 0 & -1 & -1 \\ 0 & D + \frac{1}{\theta} & -\frac{(c1 - c0)\delta}{V0} & -\frac{(c2 - c1)\delta}{V0} \end{bmatrix}$$

Let us check whether or not the system is parametrizable and flat.

```
> Ext1 := Exti(Involution(R3, Alg3), Alg3, 1);
```



$$\begin{aligned}
Ext1 := & \left[ \begin{array}{cc} 1 & 0 \\ 0 & 1 \end{array} \right], \left[ \begin{array}{cccc} 1+2\theta D & 0 & -2\theta & -2\theta \\ 0 & V0 + \theta D V0 & -\theta \delta c1 + \theta \delta c0 & \theta \delta c1 - \theta \delta c2 \end{array} \right], \\
& [-2c2\theta V0 + 4c1\theta V0 - 2c0\theta V0, -2c2\theta V0 + 4c1\theta V0 - 2c0\theta V0] \\
& [-2c2\delta c0\theta + 6c2\delta c1\theta + 2c1\delta c0\theta - 4\delta c1^2\theta - 2\delta c2^2\theta, \\
& 2c2\delta c0\theta - 2c2\delta c1\theta - 6c1\delta c0\theta + 4\delta c1^2\theta + 2\delta c0^2\theta] \\
& [-c1 V0 + c2 V0, \\
& -c2 V0 + 3c1 V0 - 2c0\theta D V0 + 4c1\theta D V0 - 2c2\theta D V0 - 2c0 V0] \\
& [3c1 V0 - c0 V0 - 2c2 V0 + 4c1\theta D V0 - 2c2\theta D V0 - 2c0\theta D V0, \\
& -c1 V0 + c0 V0]
\end{aligned}$$

The fact that  $Ext1[1]$  is the identity matrix means that the  $Alg3$ -module  $M3$  associated with the system is torsion-free, and thus, the system is parametrizable.

The parametrization of the system is given by  $Ext1[3]$  or equivalently by:

> Parametrization(R3, Alg3);

$$\begin{aligned}
& \left[ -2\xi_1(t) c2\theta V0 + 4\xi_1(t) c1\theta V0 - 2c0\theta V0 \xi_1(t) - 2\xi_2(t) c2\theta V0 \right. \\
& \left. + 4\xi_2(t) c1\theta V0 - 2c0\theta V0 \xi_2(t) \right] \\
& \left[ -2\xi_1(t-h) c0 c2\theta + 6\xi_1(t-h) c2 c1\theta + 2\xi_1(t-h) c0 c1\theta - 4c1^2\theta \xi_1(t-h) \right. \\
& - 2\xi_1(t-h) c2^2\theta + 2\xi_2(t-h) c0 c2\theta - 2\xi_2(t-h) c2 c1\theta - 6\xi_2(t-h) c0 c1\theta \\
& \left. + 4\xi_2(t-h) c1^2\theta + 2c0^2\theta \xi_2(t-h) \right] \\
& \left[ -c1 V0 \xi_1(t) + c2 V0 \xi_1(t) - c2 V0 \xi_2(t) + 3c1 V0 \xi_2(t) - 2c0\theta V0 D(\xi_2)(t) \right. \\
& \left. + 4c1\theta V0 D(\xi_2)(t) - 2c2\theta V0 D(\xi_2)(t) - 2c0 V0 \xi_2(t) \right] \\
& \left[ 3c1 V0 \xi_1(t) - c0 V0 \xi_1(t) - 2c2 V0 \xi_1(t) + 4c1\theta V0 D(\xi_1)(t) \right. \\
& \left. - 2c2\theta V0 D(\xi_1)(t) - 2c0\theta V0 D(\xi_1)(t) - c1 V0 \xi_2(t) + c0 V0 \xi_2(t) \right]
\end{aligned}$$

In order to check whether or not the system is flat, we need to investigate whether or not the module  $M3$  is projective.

> Ext2 := Exti(Involution(R3, Alg3), Alg3, 2);

$$Ext2 := \left[ \left[ \begin{array}{cc} \delta & 0 \\ \theta D + 1 & 0 \\ 0 & \delta \\ 0 & \theta D + 1 \end{array} \right], \left[ \begin{array}{cc} 1 & 0 \\ 0 & 1 \end{array} \right], SURJ(2) \right]$$

As the first matrix  $Ext2[1]$  of  $Ext2$  is not the identity matrix, the module  $M3$  is not a reflexive  $Alg3$ -module, and thus, not a free  $Alg3$ -module. Therefore, the system is not flat.

> Ext3 := Exti(Involution(R3, Alg3), Alg3, 3);

$$Ext3 := [undefined, ZERO, ZERO]$$

Therefore, the only obstruction of flatness is given by  $Ext2$ . In particular, this is coherent with the fact that the parametrization  $Ext1[3]$  is not injective as we can check as follows:

> LeftInverse(Ext1[3], Alg3);

□

Let us compute a polynomial in the time-delay operator  $\delta$  which is the obstruction of flatness.

```
> PiPolynomial(R3, Alg3, [delta]);
[delta]
```

Therefore, if we invert  $\delta$ , i.e., if we use advance operators, the system becomes flat (over the new ring  $Alg3[\delta^{-1}]$ ). Let us compute a basis:

```
> S3 := map(factor, LocalLeftInverse(Ext1[3], [delta], Alg3));
S3 := \begin{bmatrix} -\frac{-c1 + c0}{2(-2c1 + c0 + c2)^2 \theta V0} & -\frac{1}{2\delta\theta(-2c1 + c0 + c2)^2} & 0 & 0 \\ \frac{c1 - c2}{2(-2c1 + c0 + c2)^2 \theta V0} & \frac{1}{2\delta\theta(-2c1 + c0 + c2)^2} & 0 & 0 \end{bmatrix}
```

Thus, we obtain that a basis (a flat output) is given by  $(\xi_1 : \xi_2)^T = S_3 (x_1 : x_2 : u_1 : u_2)^T$ .

```
> simplify(evalm(S3 &* Ext1[3]));
\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}
> T3 := simplify(evalm(Ext1[3] &* S3));
T3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ -\frac{(1 + 2\theta D)(c1 - c2)}{2\theta(-2c1 + c0 + c2)} & -\frac{(\theta D + 1)V0}{(-2c1 + c0 + c2)\delta\theta} & 0 & 0 \\ \frac{(1 + 2\theta D)(-c1 + c0)}{2\theta(-2c1 + c0 + c2)} & \frac{(\theta D + 1)V0}{(-2c1 + c0 + c2)\delta\theta} & 0 & 0 \end{bmatrix}
```

The previous computation also shows that we can take  $(x_1 : x_2)$  as a  $\delta$ -flat output of the system. The third and fourth rows of  $T_3$  express the inputs  $u_1$  and  $u_2$  in terms of  $x_1$  and  $x_2$ . More precisely, we have:

```
> P3 := ApplyMatrix(T3, [x1(t), x2(t), u1(t), u2(t)], Alg3):
> evalm([[x1(t)], [x2(t)], [u1(t)], [u2(t)]])=eval(P3);
```

$$\begin{bmatrix} x1(t) \\ x2(t) \\ u1(t) \\ u2(t) \end{bmatrix} = \begin{bmatrix} x1(t) \\ x2(t) \\ -\frac{1}{2} \frac{(c1 - c2)x1(t)}{\theta(-2c1 + c0 + c2)} - \frac{1}{2} \frac{(-2c2 + 2c1)D(x1)(t)}{-2c1 + c0 + c2} - \frac{V0x2(t+h)}{(-2c1 + c0 + c2)\theta} \\ -\frac{V0D(x2)(t+h)}{-2c1 + c0 + c2} \\ \frac{1}{2} \frac{(-c1 + c0)x1(t)}{\theta(-2c1 + c0 + c2)} + \frac{1}{2} \frac{(-2c1 + 2c0)D(x1)(t)}{-2c1 + c0 + c2} + \frac{V0x2(t+h)}{(-2c1 + c0 + c2)\theta} \\ + \frac{V0D(x2)(t+h)}{-2c1 + c0 + c2} \end{bmatrix}$$

Let us notice that the  $\delta$ -flat output  $(x1 : x2)$  exists if and only if the following relation on the coefficients is satisfied:

$$c1 \neq \frac{c0 + c2}{2}.$$

```
> u1ref3 := t -> evalm(P3[3,1]): u1ref3(t);
```


$$-\frac{1}{2} \frac{(c1 - c2) x1(t)}{\theta(-2c1 + c0 + c2)} - \frac{1}{2} \frac{(-2c2 + 2c1) D(x1)(t)}{-2c1 + c0 + c2} - \frac{V0 x2(t+h)}{(-2c1 + c0 + c2) \theta} - \frac{V0 D(x2)(t+h)}{-2c1 + c0 + c2}$$

```
> u2ref3:= t-> evalm(P3[4,1]): u2ref3(t);
```

$$\frac{1}{2} \frac{(-c1 + c0) x1(t)}{\theta(-2c1 + c0 + c2)} + \frac{1}{2} \frac{(-2c1 + 2c0) D(x1)(t)}{-2c1 + c0 + c2} + \frac{V0 x2(t+h)}{(-2c1 + c0 + c2) \theta} + \frac{V0 D(x2)(t+h)}{-2c1 + c0 + c2}$$

Finally, if we give the trajectories  $x1ref(t)$  and  $x2ref(t)$  that we want to follow, then, by substitution into the previous equations, we obtain the corresponding open-loop inputs  $u1ref(t)$  and  $u2ref(t)$  (motion planning). Let us give an example.

```
> x1ref3 := 0.1*exp(-alpha*t): x2ref3 := 0.2*exp(-beta*t):
> alpha := 1/10: beta := 1/7:
> plot([x1ref3,x2ref3], t=0..70, color=[red,green]);
```



Let us choose a time-delay of length  $h = 5$ . Then, the corresponding open-loop inputs are defined by:


```
> u1ref5 := subs(h=5, c1=1,c2=2,c0=1.25,V0=1,theta=50,
> ApplyMatrix(linalg[submatrix](T3,3..4, 1..2), [x1ref3,x2ref3], Alg3)[1,1]);
```

$$u1ref5 := -0.007200000000 e^{(-\frac{t}{10})} + 0.01965714286 e^{(-\frac{t}{7}-5/7)}$$

```

> u2ref5 := subs(h=5,c1=1,c2=2,c0=1.25,V0=1,theta=50,
> ApplyMatrix(linalg[submatrix](T3,3..4, 1..2), [x1ref3,x2ref3], Alg3)[2,1]);
      u2ref5 := -0.001800000000 e(-t/10) - 0.01965714286 e(-t/7-5/7)
> plot([u1ref5,u2ref5], t=0..70, color=[blue,magenta]);

```




Let us choose a time-delay of length  $h = 10$ . Then, the corresponding open-loop inputs are defined by:

```

> u1ref10 := subs([h=10, c1=1,c2=2,c0=1.25,V0=1,theta=50],
> ApplyMatrix(linalg[submatrix](T3,3..4, 1..2), [x1ref3,x2ref3], Alg3)[1,1]);
      u1ref10 := -0.007200000000 e(-t/10) + 0.01965714286 e(-t/7-10/7)
> u2ref10 := subs([h=10,c1=1,c2=2,c0=1.25,V0=1,theta=50],
> ApplyMatrix(linalg[submatrix](T3,3..4, 1..2), [x1ref3,x2ref3], Alg3)[2,1]);
      u2ref10 := -0.001800000000 e(-t/10) - 0.01965714286 e(-t/7-10/7)
> plot([u1ref10,u2ref10], t=0..70, color=[blue,magenta]);

```



Previously, we saw that the system was flat if  $c1 \neq \frac{c0+c2}{2}$ . Let us study the case where  $c1 = \frac{c0+c2}{2}$ . Then, the matrix of the system becomes:

```
> R3mod := simplify(subs(c1=(c0+c2)/2, evalm(R3)));
```

$$R3mod := \begin{bmatrix} \frac{1+2\theta D}{2\theta} & 0 & -1 & -1 \\ 0 & \frac{\theta D+1}{\theta} & \frac{(-c2+c0)\delta}{2V0} & \frac{(-c2+c0)\delta}{2V0} \end{bmatrix}$$

Let us study whether or not the new system is controllable, parametrizable and flat.

```
> Ext1mod := Exti(Involution(R3mod, Alg3), Alg3, 1);
```

$$Ext1mod := \left[ \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 1+2\theta D & 0 & -2\theta & -2\theta \\ 0 & 2\theta D V0 + 2V0 & -\theta\delta c2 + \theta\delta c0 & -\theta\delta c2 + \theta\delta c0 \end{bmatrix}, \right. \\ \left. \begin{bmatrix} 4\theta^2 D V0 + 4\theta V0 & 0 \\ \theta\delta c2 - \theta\delta c0 + 2\theta^2 D\delta c2 - 2\theta^2 D\delta c0 & 0 \\ 0 & 1 \\ 6\theta D V0 + 4\theta^2 D^2 V0 + 2V0 & -1 \end{bmatrix} \right]$$

We obtain that the system is controllable, and thus, parametrizable. Then,  $Ext1mod[3]$  is a parametrization of the system and we have:

```
> evalm([[x1(t)], [x2(t)], [u1(t)], [u2(t)]] = Parametrization(R3mod, Alg3);
```

$$\begin{bmatrix} x1(t) \\ x2(t) \\ u1(t) \\ u2(t) \end{bmatrix} = \begin{bmatrix} 4\theta^2 V0 D(\xi_1)(t) + 4\theta V0 \xi_1(t) \\ c2\theta \xi_1(t-h) - \theta c0 \xi_1(t-h) + 2\theta^2 c2 D(\xi_1)(t-h) - 2\theta^2 c0 D(\xi_1)(t-h) \\ \xi_2(t) \\ 6\theta V0 D(\xi_1)(t) + 4\theta^2 V0 (D^{(2)})(\xi_1)(t) + 2V0 \xi_1(t) - \xi_2(t) \end{bmatrix}$$

Let us check whether or not the system is flat.

> Exti(Involution(R3mod, Alg3), Alg3, 2);

$$\left[ \left[ \begin{array}{cc} \delta & 0 \\ \theta D + 1 & 0 \\ 0 & 1 \end{array} \right], \left[ \begin{array}{cc} 1 & 0 \\ 0 & 1 \end{array} \right], \text{SURJ}(2) \right]$$

As the first matrix is not the identity matrix, we obtain that  $\text{ext}^2$  of the  $Alg$ -module associated with  $R3mod_{adj}$  is not the zero module, and thus, the  $Alg$ -module  $M3mod$  associated with  $R3mod$  is not free. Therefore, the system is not flat. In particular, we know that the parametrization  $Ext1mod[3]$  is not injective. Let us verify this fact.

> LeftInverse(Ext1mod[3], Alg3);

□

Moreover, as the matrix  $R3mod$  has full row rank as we have

> SyzygyModule(R3mod, Alg3);

INJ(2)

then, we also know that there is no right-inverse of  $R3mod$ . This fact can be easily checked as follows:

> RightInverse(R3mod, Alg3);

□

Therefore, the torsion-free degree of the system equals 1. Hence, there exists a unique monic polynomial  $\pi$  in  $\delta$  which corresponds to the obstruction for the system to be flat. Let us compute this polynomial.

> PiPolynomial(R3mod, Alg3, [delta]);

[ $\delta$ ]

Therefore,  $\pi = \delta$  and the system is called  $\delta$ -free or  $\delta$ -flat. In particular, this result means that there exists an injective parametrization the system if we allow the ordinary differential operator  $D$ , the time-delay operator  $\delta$  and the advance operator  $\delta^{-1}$ . Let us compute such an injective parametrization.

> S3mod := LocalLeftInverse(Ext1mod[3], [delta], Alg3);

$$S3mod := \begin{bmatrix} \frac{1}{2\theta V0} & \frac{1}{\delta\theta(-c2+c0)} & 0 & 0 \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

> simplify(evalm(S3mod &\* Ext1mod[3]));

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Therefore, if we also use the advance operator  $\delta^{-1}$ , then the flat output  $(\xi_1 : \xi_2)^T$  of the system is defined by

> evalm([[xi[1](t)], [xi[2](t)]])=  
> ApplyMatrix(S3mod, [x1(t), x2(t), u1(t), u2(t)], Alg3);

$$\begin{bmatrix} \xi_1(t) \\ \xi_2(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2} \frac{x1(t)}{\theta V0} + \frac{x2(t+h)}{\theta(-c2+c0)} \\ u1(t) \end{bmatrix}$$

and it satisfies  $(x1 : x2 : u1 : u2)^T = Ext1mod[3] (\xi_1 : \xi_2)^T$ . Let us notice that this flat output exists if and only if  $c2 \neq c0$ .

Finally, if  $c1 = \frac{c0+c2}{2}$  and  $c2 = c0$ , i.e.,  $c0 = c1 = c2$ , then the matrix of the system becomes:

```
> subs([c2=c0,c1=c0], evalm(R3));
```

$$\begin{bmatrix} D + \frac{1}{2\theta} & 0 & -1 & -1 \\ 0 & D + \frac{1}{\theta} & 0 & 0 \end{bmatrix}$$

We find again the case which occurred in the study of the ordinary differential model (the system has no time-delay operator).

To finish, let us choose the same outputs as in the ordinary differential model, i.e.:

```
> Ry := evalm([[1/(2*theta),0], [0,1]]);
```

$$Ry := \begin{bmatrix} \frac{1}{2\theta} & 0 \\ 0 & 1 \end{bmatrix}$$

Let us compute the input-output behaviour of the system. In order to do that, we first introduce the matrix  $Rt$  which is defined by  $Rt = (R^T : (Ry : 0)^T)^T$ .

```
> Rt := linalg[stackmatrix](R3, linalg[augment](Ry, matrix(2,2,0)));
```

$$Rt := \begin{bmatrix} D + \frac{1}{2\theta} & 0 & -1 & -1 \\ 0 & D + \frac{1}{\theta} & -\frac{(c1 - c0)\delta}{V0} & -\frac{(c2 - c1)\delta}{V0} \\ \frac{1}{2\theta} & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix}$$

Therefore, the system can be written as  $Rt (x1 : x2 : u1 : u2)^T = (0 : 0 : y1 : y2)$ . Now, we try to eliminate  $x1$  and  $x2$  in order to obtain equations in  $y_i$  and  $u_j$  only.

```
> E3 := Elimination(Rt, [x1,x2,u1,u2], [0,0,y1,y2], Alg3, [u1,u2]);
```

$$E3 := \text{table}([1 = \begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 0 & 1 \\ 1 & 0 \end{bmatrix},$$

$$2 = \begin{bmatrix} 0 & -\theta D V0 - V0 & \theta \delta c1 - \theta \delta c0 & -\theta \delta c1 + \theta \delta c2 \\ -2\theta D - 1 & 0 & 1 & 1 \\ 0 & 1 & 0 & 0 \\ 2\theta & 0 & 0 & 0 \end{bmatrix}$$

$$])$$

Finally, we get:

```
> ApplyMatrix(E3[1], [x1(t),x2(t)], Alg3)=
> ApplyMatrix(E3[2], [y1(t),y2(t),u1(t),u2(t)], Alg3);
```

$$\begin{bmatrix} 0 \\ 0 \\ x2(t) \\ x1(t) \end{bmatrix} = \begin{bmatrix} -\theta V0 D(y2)(t) - V0 y2(t) + c1 \theta u1(t-h) - \theta c0 u1(t-h) - c1 \theta u2(t-h) \\ + c2 \theta u2(t-h) \\ -2\theta D(y1)(t) - y1(t) + u1(t) + u2(t) \\ y2(t) \\ 2\theta y1(t) \end{bmatrix}$$

The first two equations give the input-output behaviour of the system, i.e.:

```
> IO3 := {seq(ApplyMatrix(linalg[submatrix](E3[2],1..2,1..4),
> [y1(t),y2(t),u1(t),u2(t)], Alg3)[i,1]=0, i=1..2)};
IO3 := {-2\theta D(y1)(t) - y1(t) + u1(t) + u2(t) = 0, -\theta V0 D(y2)(t) - V0 y2(t)
+ c1 \theta u1(t-h) - \theta c0 u1(t-h) - c1 \theta u2(t-h) + c2 \theta u2(t-h) = 0}
```

To finish, let us also notice that the state variables  $x1$  and  $x2$  can be expressed without derivatives in terms of  $y_i$  (and  $u_j$ ), and thus, the system is observable.

## 4 Stirred tank: a discrete-time model with a time-delay multiple of the sampling period

We now consider the discretization of the differential time-delay model of the stirred tank in the case where the delay time  $h$  is a multiple of the sampling period  $\Delta$ . For sake of simplicity, we assume  $h = \Delta$ .

See H. Kwakernaak & R. Sivan, *Linear Optimal Control Systems*, Wiley-Interscience, 1972, p. 450-451.

```
> Alg4 := DefineOreAlgebra(shift=[sigma,i], polynom=[i],
> comm=[theta,V0,c0,c1,c2,kappa]);
```

Then, the system has the form  $x(i+1) = F x(i) + G1 u(i) + G2 u(i-1)$ , where the matrices  $F$ ,  $G1$  and  $G2$  are defined by:

```
> F := evalm([[kappa,0],[0,kappa^2]]);
F := \begin{bmatrix} \kappa & 0 \\ 0 & \kappa^2 \end{bmatrix}
> G1 := evalm([[2*theta*(1-kappa),2*theta*(1-kappa)],[0,0]]);
G1 := \begin{bmatrix} 2\theta(1-\kappa) & 2\theta(1-\kappa) \\ 0 & 0 \end{bmatrix}
> G2 := evalm([[0,0],[theta*(c1-c0)*(1-kappa^2)/V0,theta*(c2-c0)*(1-kappa^2)/V0]]);
G2 := \begin{bmatrix} 0 & 0 \\ \frac{\theta(c1-c0)(1-\kappa^2)}{V0} & \frac{\theta(c2-c0)(1-\kappa^2)}{V0} \end{bmatrix}
```

It is quite common to rewrite the system  $x(i+1) = F x(i) + G1 u(i) + G2 u(i-1)$  in the form

$$X(i+1) = AX(i) + BU(i)$$



by introducing the new variables  $v1(i+1) = u1(i)$  and  $v2(i+1) = u2(i)$ . See J. Ackermann, *Sampled-Data Control Systems*, Springer-Verlag, 1985. Let us compute the system matrix of the last system.

```
> U1 := linalg[augment](sigma-F, -G2, -G1);
> U2 := linalg[augment]([[0,0],[0,0]], [[sigma,0],[0,sigma]], [[-1,0],[0,-1]]);
> R4 := linalg[stackmatrix](U1, U2);
```

$$R_4 := \begin{bmatrix} -\kappa + \sigma, 0, 0, 0, -2\theta(1-\kappa), -2\theta(1-\kappa) \\ 0, -\kappa^2 + \sigma, -\frac{\theta(c1-c0)(1-\kappa^2)}{V0}, -\frac{\theta(c2-c0)(1-\kappa^2)}{V0}, 0, 0 \\ 0, 0, \sigma, 0, -1, 0 \\ 0, 0, 0, \sigma, 0, -1 \end{bmatrix}$$

Hence, the equations of the system are defined by:

```
> ApplyMatrix(R4, [x1(i),x2(i),v1(i),v2(i),u1(i),u2(i)], Alg4)=evalm([[0]$4]);
```

$$\begin{bmatrix} -\kappa x1(i) + x1(i+1) - 2\theta(1-\kappa)u1(i) - 2\theta(1-\kappa)u2(i) \\ -\kappa^2 x2(i) + x2(i+1) - \frac{\theta(c1-c0)(1-\kappa^2)v1(i)}{V0} - \frac{\theta(c2-c0)(1-\kappa^2)v2(i)}{V0} \\ v1(i+1) - u1(i) \\ v2(i+1) - u2(i) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

Let us compute the rank of the  $Alg_4$ -module associated with  $R_4$ .

```
> OreRank(R4, Alg4);
```

2

This number corresponds to the number of independent inputs, i.e.,  $u1$  and  $u2$ . Let us check whether or not the system is controllable, parametrizable and flat. In order to do that, we first compute an adjoint  $R_4\_adj$  of  $R_4$  by using an involution of  $Alg_4$ .

```
> R4_adj := Involution(R4, Alg4);
```

$$R_4\_adj := \begin{bmatrix} -\kappa - \sigma & 0 & 0 & 0 \\ 0 & -\kappa^2 - \sigma & 0 & 0 \\ 0 & -\frac{\theta c1}{V0} + \frac{\theta c1 \kappa^2}{V0} + \frac{\theta c0}{V0} - \frac{\theta c0 \kappa^2}{V0} & -\sigma & 0 \\ 0 & -\frac{\theta c2}{V0} + \frac{\theta c2 \kappa^2}{V0} + \frac{\theta c0}{V0} - \frac{\theta c0 \kappa^2}{V0} & 0 & -\sigma \\ -2\theta + 2\theta \kappa & 0 & -1 & 0 \\ -2\theta + 2\theta \kappa & 0 & 0 & -1 \end{bmatrix}$$

Then, the system is controllable if and only if the  $Alg_4$ -module associated with  $R_4$  is torsion-free, i.e., if and only if  $ext^1$  of the  $Alg_4$ -module associated with  $R_4\_adj$  is the zero module. Let us check this last point.

```
> Ext4 := Exti(R4_adj, Alg4, 1);
```

$$\begin{aligned}
Ext4 &:= \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \\
&[-\kappa + \sigma, 0, 0, 0, -2\theta + 2\theta\kappa, -2\theta + 2\theta\kappa] \\
&\left[ 0, -V0\kappa^2 + \sigma V0, -\theta\kappa^2 c0 + c1\theta\kappa^2 + c0\theta - \theta c1, \right. \\
&\left. -\theta\kappa^2 c0 + \theta\kappa^2 c2 - \theta c2 + c0\theta, 0, 0 \right] \\
&[0, 0, \sigma, 0, -1, 0] \\
&[0, 0, 0, \sigma, 0, -1], \\
&[2V0\sigma c1\theta - 2V0\sigma c1\theta\kappa - 2V0\sigma c2\theta + 2V0\sigma c2\theta\kappa, \\
&2V0\sigma c1\theta - 2V0\sigma c1\theta\kappa - 2V0\sigma c2\theta + 2V0\sigma c2\theta\kappa] \\
&\left[ -\theta c2^2 - c1\theta c2\kappa^2 + c1\theta c0\kappa^2 - c0\theta c2\kappa^2 + c1\theta c2 - c1\theta c0 + c0\theta c2 \right. \\
&\left. + \theta c2^2\kappa^2, -c1^2\theta\kappa^2 + c1^2\theta + c1\theta c2\kappa^2 + c1\theta c0\kappa^2 - c0\theta c2\kappa^2 - c1\theta c2 \right. \\
&\left. - c1\theta c0 + c0\theta c2 \right] \\
&\left[ -V0\kappa^2 c2 + V0\kappa^2 c0 + V0\kappa c2 - V0\kappa c0, \right. \\
&\left. c1\sigma V0 - c1V0\kappa^2 + V0\kappa^2 c0 - c2\sigma V0 + V0\kappa c2 - V0\kappa c0 \right] \\
&\left[ c1\sigma V0 - V0\kappa^2 c0 - c2\sigma V0 + V0\kappa^2 c2 - \kappa c1 V0 + V0\kappa c0, \right. \\
&\left. c1V0\kappa^2 - V0\kappa^2 c0 - \kappa c1 V0 + V0\kappa c0 \right] \\
&\left[ V0\kappa\sigma c2 - V0\kappa\sigma c0 - V0\kappa^2\sigma c2 + \%1, \right. \\
&\left. V0\kappa\sigma c2 - V0\kappa\sigma c0 - c1V0\kappa^2\sigma + V0c1\sigma^2 - V0\sigma^2 c2 + \%1 \right] \\
&\left[ -V0\sigma^2 c2 + V0\kappa^2\sigma c2 - \%1 - V0\sigma c1\kappa + V0\kappa\sigma c0 + V0c1\sigma^2, \right. \\
&\left. c1V0\kappa^2\sigma - \%1 - V0\sigma c1\kappa + V0\kappa\sigma c0 \right] \\
&\%1 := V0\kappa^2\sigma c0
\end{aligned}$$

As the first matrix  $Ext4[1]$  is the identity matrix, we obtain that the system is controllable, and thus, parametrizable. A parametrization of the system is then defined by  $Ext4[3]$ . Equivalently, we have:

```

> evalm([[x1(i)], [x2(i)], [v1(i)], [v2(i)], [u1(i)], [u2(i)]]]=
> Parametrization(R4, Alg4);

```

$$\begin{aligned}
& \begin{bmatrix} x1(i) \\ x2(i) \\ v1(i) \\ v2(i) \\ u1(i) \\ u2(i) \end{bmatrix} = \\
& \left[ 2\xi_1(i+1) V0 c1 \theta - 2\xi_1(i+1) V0 c1 \theta \kappa - 2\xi_1(i+1) V0 c2 \theta \right. \\
& \quad + 2\xi_1(i+1) V0 c2 \theta \kappa + 2\xi_2(i+1) V0 c1 \theta - 2\xi_2(i+1) V0 c1 \theta \kappa \\
& \quad \left. - 2\xi_2(i+1) V0 c2 \theta + 2\xi_2(i+1) V0 c2 \theta \kappa \right] \\
& \left[ -\xi_1(i) \theta c2^2 - \xi_1(i) c1 \theta c2 \kappa^2 + \xi_1(i) c1 \theta c0 \kappa^2 - \xi_1(i) c0 \theta c2 \kappa^2 + \xi_1(i) c1 \theta c2 \right. \\
& \quad - \xi_1(i) c1 \theta c0 + \xi_1(i) c0 \theta c2 + \xi_1(i) \theta c2^2 \kappa^2 - \xi_2(i) c1^2 \theta \kappa^2 + \xi_2(i) c1^2 \theta \\
& \quad + \xi_2(i) c1 \theta c2 \kappa^2 + \xi_2(i) c1 \theta c0 \kappa^2 - \xi_2(i) c0 \theta c2 \kappa^2 - \xi_2(i) c1 \theta c2 \\
& \quad \left. - \xi_2(i) c1 \theta c0 + \xi_2(i) c0 \theta c2 \right] \\
& \left[ -\xi_1(i) V0 \kappa^2 c2 + \xi_1(i) V0 \kappa^2 c0 + \xi_1(i) V0 \kappa c2 - \xi_1(i) V0 \kappa c0 + \xi_2(i) V0 \kappa c2 \right. \\
& \quad - \xi_2(i) c1 V0 \kappa^2 + \xi_2(i) V0 \kappa^2 c0 - \xi_2(i) V0 \kappa c0 - \xi_2(i+1) c2 V0 \\
& \quad \left. + \xi_2(i+1) c1 V0 \right] \\
& \left[ -\xi_1(i) \kappa c1 V0 - \xi_1(i) V0 \kappa^2 c0 + \xi_1(i) V0 \kappa^2 c2 + \xi_1(i) V0 \kappa c0 - \xi_1(i+1) c2 V0 \right. \\
& \quad + \xi_1(i+1) c1 V0 + \xi_2(i) c1 V0 \kappa^2 - \xi_2(i) V0 \kappa^2 c0 - \xi_2(i) \kappa c1 V0 + \xi_2(i) V0 \kappa c0 \left. \right] \\
& \left[ -\xi_1(i+1) V0 \kappa^2 c2 + \xi_1(i+1) V0 \kappa^2 c0 + \xi_1(i+1) V0 \kappa c2 - \xi_1(i+1) V0 \kappa c0 \right. \\
& \quad - \xi_2(i+2) c2 V0 + \xi_2(i+2) c1 V0 + \xi_2(i+1) V0 \kappa c2 - \xi_2(i+1) c1 V0 \kappa^2 \\
& \quad \left. + \xi_2(i+1) V0 \kappa^2 c0 - \xi_2(i+1) V0 \kappa c0 \right] \\
& \left[ -\xi_1(i+2) c2 V0 + \xi_1(i+2) c1 V0 - \xi_1(i+1) \kappa c1 V0 - \xi_1(i+1) V0 \kappa^2 c0 \right. \\
& \quad + \xi_1(i+1) V0 \kappa^2 c2 + \xi_1(i+1) V0 \kappa c0 + \xi_2(i+1) c1 V0 \kappa^2 - \xi_2(i+1) V0 \kappa^2 c0 \\
& \quad \left. - \xi_2(i+1) \kappa c1 V0 + \xi_2(i+1) V0 \kappa c0 \right]
\end{aligned}$$

Let us point out that  $Alg_4$  is a hereditary ring, and thus, the  $Alg_4$ -module associated with  $R_4$  is projective. This last fact can be directly verified by checking the existence of a right-inverse of  $R_4$ .

> T4 := simplify(RightInverse(R4, Alg4));

$$T_4 := \begin{bmatrix} -\frac{1}{\kappa} & 0 & -\frac{2\theta(-1+\kappa)}{\kappa} & -\frac{2\theta(-1+\kappa)}{\kappa} \\ 0 & 0 & 0 & 0 \\ \frac{-c_2+c_0}{2\theta\kappa\%2} & \frac{V_0}{\%1\theta} & \frac{-c_2+c_0}{\kappa(c_1-c_2)} & \frac{-c_2+c_0}{\kappa(c_1-c_2)} \\ -\frac{-c_1+c_0}{2\theta\kappa\%2} & -\frac{V_0}{\%1\theta} & -\frac{-c_1+c_0}{\kappa(c_1-c_2)} & -\frac{-c_1+c_0}{\kappa(c_1-c_2)} \\ \frac{\sigma(-c_2+c_0)}{2\theta\kappa\%2} & \frac{V_0\sigma}{\%1\theta} & \frac{-\sigma c_2+\sigma c_0-c_1\kappa+c_2\kappa}{\kappa(c_1-c_2)} & \frac{\sigma(-c_2+c_0)}{\kappa(c_1-c_2)} \\ -\frac{\sigma(-c_1+c_0)}{2\theta\kappa\%2} & -\frac{V_0\sigma}{\%1\theta} & -\frac{\sigma(-c_1+c_0)}{\kappa(c_1-c_2)} & -\frac{-c_1\sigma+c_1\kappa-c_2\kappa+\sigma c_0}{\kappa(c_1-c_2)} \end{bmatrix}$$

$$\%1 := c_1\kappa^2 - c_1 - c_2\kappa^2 + c_2$$

$$\%2 := -c_1 + c_1\kappa + c_2 - c_2\kappa$$

We check that  $R_4 T_4 = I$ :

> evalm(Mult(R4, T4, Alg4));

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Finally, the fact that the system is time-invariant implies that the  $Alg_4$ -module associated with  $R_4$  is free. Therefore, the system is flat and a flat output of the system is defined by means of a left-inverse of the parametrization  $Ext_4$  [3].

> S4 := simplify(LeftInverse(Ext4[3], Alg4));

$$S_4 := \begin{bmatrix} \frac{-c_1+c_0}{2(-2c_1\kappa c_2+\kappa c_2^2+c_1^2\kappa+2c_2c_1-c_1^2-c_2^2)V_0\kappa\theta}, \\ \frac{1}{\theta(c_1^2\kappa^2-c_1^2-2c_1c_2\kappa^2+2c_2c_1-c_2^2+c_2^2\kappa^2)}, \frac{-c_1+c_0}{\%1V_0\kappa}, \frac{-c_1+c_0}{\%1V_0\kappa}, 0, 0 \\ -\frac{-c_2+c_0}{2(-2c_1\kappa c_2+\kappa c_2^2+c_1^2\kappa+2c_2c_1-c_1^2-c_2^2)V_0\kappa\theta}, \\ -\frac{1}{\theta(c_1^2\kappa^2-c_1^2-2c_1c_2\kappa^2+2c_2c_1-c_2^2+c_2^2\kappa^2)}, -\frac{-c_2+c_0}{\%1V_0\kappa}, -\frac{-c_2+c_0}{\%1V_0\kappa}, \\ 0, 0 \end{bmatrix}$$

$$\%1 := -2c_2c_1 + c_2^2 + c_1^2$$

Hence, a flat output of the system is defined by  $\xi[1] = S_4 (x_1 : x_2 : v_1 : v_2 : u_1 : u_2)^T$ , namely:

> xi[1](i)=ApplyMatrix(S4, [x1(i),x2(i),v1(i),v2(i),u1(i),u2(i)], Alg4)[1,1];

$$\begin{aligned} \xi_1(i) &= \frac{1}{2} \frac{(-c_1+c_0)x_1(i)}{(-2c_1\kappa c_2+\kappa c_2^2+c_1^2\kappa+2c_2c_1-c_1^2-c_2^2)V_0\kappa\theta} \\ &+ \frac{x_2(i)}{\theta(c_1^2\kappa^2-c_1^2-2c_1c_2\kappa^2+2c_2c_1-c_2^2+c_2^2\kappa^2)} \\ &+ \frac{(-c_1+c_0)v_1(i)}{(-2c_2c_1+c_2^2+c_1^2)V_0\kappa} + \frac{(-c_1+c_0)v_2(i)}{(-2c_2c_1+c_2^2+c_1^2)V_0\kappa} \end{aligned}$$

The flat output satisfies  $(x1 : x2 : v1 : v2 : u1 : u2)^T = S_4 \xi[1]$ . Let us notice that the flat output only exists if the denominators of  $S_4$  are non-zero, i.e., if the following polynomial does not vanish:

$$\begin{aligned} > \text{factor}(\text{denom}(S4[1,1])); \\ & 2(c1 - c2)^2(-1 + \kappa) V0 \kappa \theta \end{aligned}$$

Therefore, the system is flat if  $c2 \neq c1$ . Let us consider the case where  $c2 = c1$ . Then, as in the ordinary differential model, we have  $c2 = c1 = c\theta$  and the matrix of the system becomes:

$$\begin{aligned} > R4bis := \text{subs}([c2=c0, c1=c0], \text{evalm}(R4)); \\ R4bis := & \begin{bmatrix} -\kappa + \sigma & 0 & 0 & 0 & -2\theta(1 - \kappa) & -2\theta(1 - \kappa) \\ 0 & -\kappa^2 + \sigma & 0 & 0 & 0 & 0 \\ 0 & 0 & \sigma & 0 & -1 & 0 \\ 0 & 0 & 0 & \sigma & 0 & -1 \end{bmatrix} \end{aligned}$$

Let us define  $R4bis\_adj$  of  $R4bis$  by using an involution of  $Alg4$ .

$$> R4bis\_adj := \text{Involution}(R4bis, Alg4):$$

Then, the new system is controllable and parametrizable if and only if the first extension module  $ext^1$  with values in  $Alg4$  of the  $Alg4$ -module associated with  $R4bis\_adj$  is 0. We have:

$$\begin{aligned} > Ext4bis := \text{Exti}(R4bis\_adj, Alg4, 1); \\ Ext4bis := & \left[ \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & -\kappa^2 + \sigma & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} -\kappa + \sigma & 0 & 0 & 0 & -2\theta + 2\theta\kappa & -2\theta + 2\theta\kappa \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & \sigma & 0 & -1 & 0 \\ 0 & 0 & 0 & \sigma & 0 & -1 \end{bmatrix}, \right. \\ & \left. \begin{bmatrix} 2\theta\sigma - 2\sigma\theta\kappa & 0 \\ 0 & 0 \\ 0 & -1 \\ -\kappa + \sigma & 1 \\ 0 & -\sigma \\ -\kappa\sigma + \sigma^2 & \sigma \end{bmatrix} \right] \end{aligned}$$

As the first matrix  $Ext4bis[1]$  is not the identity matrix, we obtain that the system is not controllable. In particular, the torsion elements of the  $Alg4$ -module associated with  $R4bis$  are defined by:

$$\begin{aligned} > \text{TorsionElements}(R4bis, [x1(i), x2(i), v1(i), v2(i), u1(i), u2(i)], Alg4); \\ & [[ -\kappa^2 \theta_2(i) + \theta_2(i+1) = 0 ], [ \theta_2(i) = x2(i) ]] \end{aligned}$$

The controllable subsystem is defined by  $Ext4bis[2]$ , i.e., by the following equations:

$$\begin{aligned} > \text{ApplyMatrix}(Ext4bis[2], [x1(i), x2(i), v1(i), v2(i), u1(i), u2(i)], Alg4)= \\ > \text{evalm}([[0]\$4]); \\ & \begin{bmatrix} -\kappa x1(i) + x1(i+1) + (-2\theta + 2\theta\kappa)u1(i) + (-2\theta + 2\theta\kappa)u2(i) \\ x2(i) \\ v1(i+1) - u1(i) \\ v2(i+1) - u2(i) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \end{aligned}$$

Then, the parametrization of the controllable subsystem is defined by  $Ext4bis[3]$ , i.e. we have:

```
> evalm([[x1(i)], [x2(i)], [v1(i)], [v2(i)], [u1(i)], [u2(i)]])=
> ApplyMatrix(Ext4bis[3], [xi[1](i), xi[2](i)], Alg4);
```

$$\begin{bmatrix} x1(i) \\ x2(i) \\ v1(i) \\ v2(i) \\ u1(i) \\ u2(i) \end{bmatrix} = \begin{bmatrix} (2\theta - 2\theta\kappa)\xi_1(i+1) \\ 0 \\ -\xi_2(i) \\ -\kappa\xi_1(i) + \xi_1(i+1) + \xi_2(i) \\ -\xi_2(i+1) \\ -\kappa\xi_1(i+1) + \xi_1(i+2) + \xi_2(i+1) \end{bmatrix}$$

The controllable subsystem is then flat and a flat output is defined by

$$(\xi[1] : \xi[2])^T = S4bis(x1 : x2 : v1 : v2 : u1 : u2)^T,$$

where  $S4bis$  is a left-inverse of the parametrization  $Ext4bis[3]$  defined by

```
> S4bis := LeftInverse(Ext4bis[3], Alg4);
```

$$S4bis := \begin{bmatrix} 1 & 0 & -2\theta + 2\theta\kappa & -2\theta + 2\theta\kappa & 0 & 0 \\ -\frac{1}{2\theta\kappa(-1+\kappa)} & 0 & -\frac{-2\theta + 2\theta\kappa}{2\theta\kappa(-1+\kappa)} & -\frac{-2\theta + 2\theta\kappa}{2\theta\kappa(-1+\kappa)} & 0 & 0 \\ 0 & 0 & -1 & 0 & 0 & 0 \end{bmatrix}$$

or, more precisely,

```
> evalm([[xi[1](i)], [xi[2](i)]])=
> ApplyMatrix(S4bis, [x1(i), x2(i), v1(i), v2(i), u1(i), u2(i)], Alg4);
```

$$\begin{bmatrix} \xi_1(i) \\ \xi_2(i) \end{bmatrix} = \begin{bmatrix} -\frac{1}{2\theta\kappa(-1+\kappa)}x1(i) - \frac{1}{2}\frac{(-2\theta + 2\theta\kappa)v1(i)}{\theta\kappa(-1+\kappa)} - \frac{1}{2}\frac{(-2\theta + 2\theta\kappa)v2(i)}{\theta\kappa(-1+\kappa)} \\ -v1(i) \end{bmatrix}$$